

Gaetano Marano

Programmi per Amstrad

CPC 464 - CPC 664 - CPC 6128



**GRUPPO EDITORIALE
JACKSON**

DIVISIONE LIBRI

Gaetano Marano

Programmi per **Amstrad**

CPC 464 - CPC 664 - CPC 6128



GRUPPO
EDITORIALE
JACKSON
Via Rosellini, 12
20124 Milano

©Copyright per l'edizione originale
GRUPPO EDITORIALE JACKSON - Settembre 1986

SUPERVISIONE TECNICA: Mauro Risani
GRAFICA E IMPAGINAZIONE: Silvano Teruzzi
COPERTINA: Sergio Mazzali
FOTOCOMPOSIZIONE: Corponove (BG)
STAMPA: Stabilimento Grafico A. Matarrelli - Milano

Tutti i diritti sono riservati. Stampato in Italia. Nessuna parte di questo libro può essere riprodotta, memorizzata in sistemi di archivio, o trasmessa in qualsiasi forma o mezzo, elettronico, meccanico, fotocopia, registrazione o altri senza la preventiva autorizzazione scritta dell'editore.

PREFAZIONE

Molti personal computer hanno un ottimo prezzo in rapporto alle prestazioni (che poi è la condizione principale per la loro diffusione) ma gli **Amstrad** sono in assoluto i computer ad 8 bit con il rapporto più favorevole tanto da essere stati i primi home computer ad essere venduti completi di monitor monocromatico o a colori.

Oltre alla loro convenienza gli Amstrad offrono anche delle prestazioni molto interessanti, e questo è anche dovuto al fatto che, essendo arrivati sul mercato da non molto, hanno potuto riassumere gran parte di quanto di meglio era stato fatto ed anche evitare alcuni degli errori commessi dai primi produttori di home computer.

Particolarmente interessante è senz'altro il **Locomotive BASIC** dei vari CPC 464, CPC 664 e CPC 6128, che, oltre ad essere tra i più completi, dispone anche di alcune caratteristiche innovative, come le istruzioni AFTER ed EVERY che consentono di offrire all'utente alcune forme di multitasking (in particolare l'elaborazione alternata di più parti del programma).

Nei programmi del libro, che sono di tipo molto vario, vengono usate appunto le molte caratteristiche di questo computer.

Gaetano Marano

USO DEI PROGRAMMI

Tutti i programmi sono registrati sulla cassetta allegata al libro con i nomi di A1, A2, eccetera, fino ad A23.

I programmi possono essere caricati in memoria tramite l'istruzione:

LOAD "nome programma"

dove 'nome programma' è rappresentato appunto da A1, A2, eccetera, come indicato nella lista nella pagina seguente.

Se si vuole caricare in memoria il primo programma che si incontra sul nastro è possibile usare l'istruzione:

LOAD "

Se si vuole che il programma giri automaticamente subito dopo essere stato caricato si può usare l'istruzione:

RUN "nome programma"

oppure soltanto

RUN "

Durante il funzionamento, il programma può essere fermato momentaneamente premendo una volta il tasto ESC (per continuare basta premere uno dei tasti), o definitivamente premendo due volte sempre il tasto ESC.

Infine se si vuole resettare il computer senza spegnerlo occorre premere insieme i tasti CTRL - SHIFT - ESC.

Programma	Nome
ISTOGRAMMI VERTICALI	A1
ISTOGRAMMI ORIZZONTALI	A2
CILINDRI 3D	A3
DESIGN CARATTERI	A4
DISEGNATORE GRAFICO	A5
PLOTTER	A6
CURVE MATEMATICHE	A7
NEW YORK	A8
DISEGNI DI LINEE	A9
ITALIA	A10
NUMERI DIGITALI	A11
LETTURA VELOCE	A12
CRONOMETRO	A13
SCRITTE PUBBLICITARIE	A14
CONTAPEZZI	A15
FISICA DEI REATTORI NUCLEARI	A16
PIANO 4-OTTAVE	A17
PROGRAMMATORE DI SUONI	A18
SIMMETRICO	A19
TIRO ALLA FUNE	A20
ROULETTE	A21
FLIPPER GIAPPONESE	A22
ROMBOSPIRALE	A23

ISTOGRAMMI VERTICALI

Il programma **Istogrammi Verticali** è il primo di tre programmi di 'Business Grafica' vale a dire della rappresentazione in forma grafica di dati economici, tecnici, scientifici, eccetera. Il programma consente di visualizzare, sotto forma di barre verticali, fino ad otto valori numerici contemporaneamente. Facendo girare il programma, il computer chiede per prima cosa di inserire il titolo del grafico, della lunghezza di massimo 80 caratteri.

Quindi chiede l'inserimento degli otto argomenti relativi ai dati e dei corrispondenti dati numerici.

Se si desidera inserire meno di otto dati, ad esempio si vuole un grafico con solo 5 istogrammi, è sufficiente premere **RETURN** durante l'inserimento dell'argomento del sesto istogramma per far passare il programma alla fase di disegno del grafico.

I valori numerici dei dati inseriti devono essere compresi tra 0 e 5000, penserà poi il programma ad usare la scala più adatta ai valori inseriti; per esempio, se il valore più alto inserito è 897, il programma disegnerà un grafico con un valore massimo di 1000.

Se i valori da inserire sono più grandi di 5000, basta indicare nel titolo del grafico un fattore di moltiplicazione (ad esempio Lire \times 1000) ed inserire dei valori numerici divisi per mille. Dopo la visualizzazione del grafico, è possibile passare all'inserimento dei dati per un nuovo grafico, premendo il tasto N.

```

1000 /      * Business Grafica *
1010 /      ISTOGRAMMI VERTICALI
1020 / -----
1030 /      di Gaetano Marano
1040 / -----
1050 /      (c) Gruppo Editoriale Jackson
1060 / -----
1070 MODE 1
1080 INK 1,26:INK 0,0:INK 2,24:INK 3,3
1090 PEN 1:PAPER 0:BORDER 0:CLS
1100 SOUND 1,0,0,0:SOUND 2,0,0,0
1110 SOUND 4,0,0,0
1120 / -----
1130 DIM D(8):DIM D$(8)
1140 FOR G=1 TO 8
1150 D(G)=0:D$(G)=" "
1160 NEXT G
1170 N=0:CLS
1180 Z$=SPACE$(40)
1190 Z$=Z$+Z$
1200 INPUT "Inserire titolo grafico (max
80      caratteri) e premere RETURN
      ";T$
1210 IF LEN(T$)>80 THEN T$=LEFT$(T$,80)
1220 IF LEN(T$)<80 THEN T$=T$+LEFT$(Z$,8
0-LEN(T$))
1230 FOR G=1 TO 8
1240 CLS
1250 PRINT "Inserire argomento istogramm
a";G;"(max 8 caratteri) e premere RETURN
o      premere solo RETURN per meno di 8
dati"
1260 PRINT
1270 INPUT R$
1280 IF R$="" THEN 1380
1290 IF LEN(R$)>8 THEN R$=LEFT$(R$,8)
1300 D$(G)=R$
1310 PRINT

```

```

1320 PRINT "Inserire valore";G;"(0..5000
)
";"e premere RETURN"
1330 PRINT
1340 INPUT Y
1350 IF Y<0 OR Y>5000 THEN GOTO 1320
1360 D(G)=Y
1370 NEXT G
1380 CLS:R=0
1390 FOR G=1 TO 8
1400 IF D(G)>R THEN R=D(G)
1410 NEXT G
1420 RESTORE
1430 FOR M=1 TO 9
1440 READ N$,S$
1450 IF VAL(N$)>R THEN 1470
1460 NEXT M
1470 N=VAL(N$)/196
1480 PEN 0:PAPER 2
1490 LOCATE 3,24:PRINT "   PREMERE N PER
UN ALTRO GRAFICO "
1500 PEN 1:PAPER 0
1510 FOR G=1 TO 8
1520 C=3+G*3
1530 A$=CHR$(G+64)
1540 LOCATE C,22:PRINT A$
1550 NEXT G
1560 LOCATE 31,9:PRINT N$
1570 LOCATE 31,15:PRINT S$
1580 LOCATE 31,21:PRINT "0"
1590 PEN 0:PAPER 1
1600 LOCATE 1,1:PRINT LEFT$(T$,40)
1610 LOCATE 1,2:PRINT RIGHT$(T$,40)
1620 PEN 1:PAPER 0
1630 FOR G=0 TO 7
1640 L=INT (G/3)+4
1650 IF G=0 OR G=3 OR G=6 THEN C=3
1660 IF G=1 OR G=4 OR G=7 THEN C=15
1670 IF G=2 OR G=5 OR G=8 THEN C=27

```

```

1680 A$=CHR$(G+65)
1690 PEN 0:PAPER 1
1700 LOCATE C,L:PRINT A$
1710 PEN 1:PAPER 0
1720 LOCATE C+2,L:PRINT D$(G+1)
1730 NEXT G
1740 FOR G=68 TO 268 STEP 20
1750 PLOT 55,G:DRAW 455,G,3
1760 IF G=68 OR G=168 OR G=268 THEN PLOT
    55,G:DRAW 475,G,3
1770 NEXT G
1780 PLOT 55,70:DRAW 55,268,3
1790 PLOT 455,70:DRAW 455,268,3
1800 FOR G=1 TO 8
1810 X=70+(G-1)*47.7
1820 V=INT(D(G)/N+0.5)
1830 FOR M=X TO X+35
1840 IF V>0 THEN PLOT M,69:DRAW M,70+V,2
1850 NEXT M
1860 NEXT G
1870 H$=INKEY$
1880 IF H$="N" OR H$="n" THEN RUN
1890 GOTO 1870
1900 RETURN
1910 DATA "10","5","20","10","50","25","
100","50","200","100","500","250","1000"
1920 DATA "500","2000","1000","5000","25
00"

```

ISTOGRAMMI ORIZZONTALI

Il funzionamento del programma **Istogrammi Orizzontali** è del tutto simile a quello del programma precedente tranne per il fatto che i dati vengono visualizzati in maniera diversa (appunto orizzontalmente).

Inoltre i dati inseriti possono essere compresi tra 0 e 6000 (invece di 5000).

```
1000 /          * Business Grafica *
1010 /          ISTOGRAMMI ORIZZONTALI
1020 / -----
1030 /          di Gaetano Marano
1040 / -----
1050 / (c) Gruppo Editoriale Jackson
1060 / -----
1070 MODE 1
1080 INK 1,26:INK 2,11:INK 0,0:INK 3 24
1090 PEN 1 BORDER 0:PAPER 0:CLS
1100 SOUND 1,0,0,0:SOUND 2,0,0,0
1110 SOUND 4,0,0,0
1120 / -----
1130 DIM D(8):DIM D$(8)
1140 FOR G=1 TO 8
1150 D(G)=0:D$(G)=""
1160 NEXT G
1170 N=0:CLS
1180 Z$=SPACE$(40)
1190 Z$=Z$+Z$
1200 INPUT "Inserire titolo grafico (max
          80 caratteri) e premere RETU
RN          ";T$
```

```

1210 IF LEN(T$)>80 THEN T$=LEFT$(T$,80)
1220 IF LEN(T$)<80 THEN T$=T$+LEFT$(Z$,80-LEN(T$))
1230 FOR G=1 TO 8
1240 CLS
1250 PRINT "Inserire argomento istogramma";G;"(max 8 caratteri) e premere RETURN
o premere solo RETURN per meno di 8
dati."
1260 PRINT
1270 INPUT R$
1280 IF R$="" THEN 1380
1290 IF LEN(R$)>8 THEN R$=LEFT$(R$,8)
1300 D$(G)=R$
1310 PRINT
1320 PRINT "Inserire valore";G;"(0-6000
)"
";"e premere RETURN"
1330 PRINT
1340 INPUT Y
1350 IF Y<0 OR Y>6000 THEN GOTO 1320
1360 D(G)=Y
1370 NEXT G
1380 CLS:R=0
1390 FOR G=1 TO 8
1400 IF D(G)>R THEN R=D(G)
1410 NEXT G
1420 RESTORE
1430 FOR M=1 TO 9
1440 READ N$,S$,Q$
1450 IF VAL(N$)>=R THEN 1470
1460 NEXT M
1470 N=VAL(N$)/475
1480 LOCATE 5,24:PEN 0:PAPER 2:PRINT "
PREMERE N PER UN'ALTRO GRAFICO ":PEN 1:
PAPER 0
1490 LOCATE 39-INT(LEN(N$)/2),21:PRINT N
$
1500 LOCATE 29-INT(LEN(S$)/2),21:PRINT S

```

```

$
1510 LOCATE 19-INT(LEN(Q$)/2),21:PRINT Q
$
1520 LOCATE 9,21:PRINT "0"
1530 PEN 0:PAPER 1
1540 LOCATE 1,1:PRINT LEFT$(T$,40)
1550 LOCATE 1,2:PRINT RIGHT$(T$,40):PEN
1
1560 PEN 1:PAPER 0
1570 L=18
1580 FOR G=1 TO 8
1590 C=9-LEN(D$(G))
1600 LOCATE C,L:PRINT D$(G)
1610 L=L-2
1620 NEXT G
1630 FOR G=135 TO 615 STEP 32
1640 PLOT G,105:DRAW G,360,3
1650 IF G=135 OR G=295 OR G=455 OR G=615
  THEN PLOT G,90:DRAW G,360,3
1660 NEXT G
1670 PLOT 135,105:DRAW 615,105,3
1680 PLOT 135,360:DRAW 615,360,3
1690 FOR G=1 TO 8
1700 Y=130+(G-1)*32
1710 V=INT(D(G)/N+0.5)
1720 FOR M=Y-22 TO Y
1730 IF V>0 THEN PLOT 136,M:DRAW 135+V,M
  ,2
1740 NEXT M
1750 NEXT G
1760 H$="A"
1770 H$=INKEY$
1780 IF H$="N" OR H$="n" THEN RUN
1790 GOTO 1770
1800 RETURN
1810 DATA "15","10","5","30","20","10","
60","40","20","150","100","50","300","20
0"

```

```
1820 DATA "100", "600", "400", "200", "1500"  
      , "1000", "500", "3000", "2000", "1000", "6000"  
      "  
1830 DATA "4000", "2000"
```


CILINDRI TRIDIMENSIONALI

L'ultimo programma di Business Grafica visualizza un massimo di sei dati sotto forma di **cilindri tridimensionali**; i dati inseriti possono essere compresi tra 0 e 9999.

```
1000 /          * Business Grafica *
1010 /          CILINDRI TRIDIMENSIONALI
1020 /          -----
1030 /          di Gaetano Marano
1040 /          -----
1050 /          (c) Gruppo Editoriale Jackson
1060 /          -----
1070 MODE 1
1080 INK 2,15:INK 0,1:INK 1,26:INK 3,18
1090 PEN 1: BORDER 1:PAPER 0:CLS
1100 SOUND 1,0,0,0:SOUND 2,0,0,0
1110 SOUND 4,0,0,0
1120 /          -----
1130 DIM D(6):DIM D$(6)
1140 FOR G=1 TO 6
1150 D(G)=0:D$(G)=" "
1160 NEXT G
1170 N=0:CLS
1180 Z$=SPACE$(40)
1190 Z$=Z$+Z$
1200 INPUT "Inserire titolo grafico (max
          80 caratteri) e premere RETU
RN          ";T$
1210 IF LEN(T$)>80 THEN T$=LEFT$(T$,80)
1220 IF LEN(T$)<80 THEN T$=T$+LEFT$(Z$,8
0-LEN(T$))
```

```

1230 FOR G=1 TO 6
1240 CLS
1250 PRINT "Inserire argomento CILINDRO"
;G;"(max 5 caratteri) e premere RETURN o
premere solo RETURN per meno di 8 da
ti"
1260 PRINT
1270 INPUT R$
1280 IF R$="" THEN 1380
1290 IF LEN(R$)>5 THEN R$=LEFT$(R$,5)
1300 D$(G)=R$
1310 PRINT
1320 PRINT "Inserire valore";G;"(0..9999
)";"e premere RETURN "
1330 PRINT
1340 INPUT Y
1350 IF Y<0 OR Y>9999 THEN GOTO 1320
1360 D(G)=Y
1370 NEXT G
1380 CLS:R=0
1390 FOR G=1 TO 6
1400 IF D(G)>R THEN R=D(G)
1410 NEXT G
1420 N=R/200
1430 PEN 0:PAPER 3
1440 LOCATE 4,24:PRINT " PREMERE N PER
UN'ALTRO GRAFICO "
1450 PEN 1:PAPER 0
1460 FOR G=1 TO 6
1470 C=G*6-3
1480 IF LEN(D$(G))<5 THEN D$(G)=D$(G)+LE
FT$(" ",5-LEN(D$(G)))
1490 LOCATE C,22:PRINT D$(G)
1500 NEXT G
1510 PEN 0:PAPER 1
1520 LOCATE 1,1:PRINT LEFT$(T$,40)
1530 LOCATE 1,2:PRINT RIGHT$(T$,40)
1540 PAPER 0:PEN 1

```

```
1550 U=0
1560 FOR G=1 TO 6
1570 IF G=1 OR G=4 THEN C=1
1580 IF G=2 OR G=5 THEN C=2
1590 IF G=3 OR G=6 THEN C=3
1600 IF D(G)=0 THEN GOTO 1760
1610 F=D(G)/N
1620 FOR P=1 TO 360
1630 X=70:Y=90
1640 AA=X-40*COS(P)
1650 BB=Y-15*SIN(P)
1660 DEG
1670 PLOT AA+U, BB+F, C
1680 IF BB>=Y THEN 1700
1690 PLOT AA+U, BB+F: DRAW AA+U, BB, C
1700 NEXT P
1710 U=U+96
1720 A$=STR$(INT(D(G)))
1730 C=G*6-4
1740 L=18-(D(G)/N+10)/15
1750 LOCATE C, L: PRINT A$
1760 NEXT G
1770 H$=INKEY$
1780 IF H$="N" OR H$="n" THEN RUN
1790 GOTO 1770
```


DESIGN CARATTERI

I caratteri dell'Amstrad sono ridefinibili tramite le istruzioni **SYMBOL** e **SYMBOL AFTER** (per il loro uso vedere il manuale) per cui è possibile creare i simboli che occorrono nel programma anche se questi non sono tra quelli disponibili nel set di caratteri del computer, inoltre, usando più caratteri definiti, si possono realizzare delle figure complesse senza dovere usare una lunga serie di istruzioni grafiche.

Per semplificare il disegno dei caratteri definibili si può usare il programma 'Design Caratteri' che consente di disegnare, modificare e stampare uno o più caratteri definiti, aventi una matrice di 8×8 punti e calcolare i codici da inserire nelle istruzioni **SYMBOL**.

Il disegno del **nuovo carattere** si può realizzare usando i quattro tasti di controllo cursore; oltre a questi tasti si può usare il tasto M per spostarsi senza disegnare o per fare delle modifiche, il tasto D per tornare al modo disegno, il tasto X per avere i codici del carattere disegnato fino a quel momento, il tasto C per avere la copia su stampante del carattere e dei codici ed il tasto CLR che annulla tutto e fa ripartire il programma.

Il tasto C va premuto, ovviamente, se al computer è collegata una stampante.

```
1000 /      ** DESIGN CARATTERI **
1010 / -----
1020 /      di Gaetano Marano
1030 / -----
1040 /      (c) Gruppo Editoriale Jackson
1050 / -----
1060 MODE 1:INK 0,0:INK 1,26
```

```

1070 INK 2,11:INK 3,6
1080 PEN 1:PAPER 0:BORDER 0:CLS
1090 SOUND 1,0,0,0:SOUND 2,0,0,0
1100 SOUND 4,0,0,0
1110 / -----
1120 C$=CHR$(150):D$=CHR$(154):E$=CHR$(1
56):F$=CHR$(149):I$=CHR$(147):M$=CHR$(15
3)
1130 N$=CHR$(144):O$=CHR$(143):P$=CHR$(2
46):Q$=CHR$(247):R$=CHR$(244):S$=CHR$(24
5)
1140 FOR W=1 TO 3
1150 N$=N$+N$:D$=D$+D$
1160 NEXT W
1170 DIM T(8,8)
1180 DIM S(8)
1190 L=2:C=15:D=1
1200 LOCATE 14,1:PRINT C$;D$;E$
1210 FOR A=2 TO 9
1220 LOCATE 14,A:PRINT F$;N$;F$
1230 NEXT A
1240 LOCATE 14,10:PRINT I$;D$;M$
1250 PEN 2
1260 LOCATE 5,14:PRINT P$;" ";Q$;" ";R$;
" ";S$;" -- Tasti per disegnare"
1270 LOCATE 5,15:PRINT "D_Disegno M_Mod
ifich e C_Copy"
1280 LOCATE 5,16:PRINT "X_Dati CLR_N
uovo carattere"
1290 PEN 1
1300 GOSUB 1560
1310 K$=INKEY$
1320 IF K$<>" " THEN K=ASC(K$)
1330 IF K=16 THEN RUN
1340 IF K=240 THEN GOSUB 1460
1350 IF K=241 THEN GOSUB 1480
1360 IF K=242 THEN GOSUB 1500
1370 IF K=243 THEN GOSUB 1520

```

```

1380 IF K=68 OR K=100 THEN GOSUB 1540
1390 IF K=77 OR K=109 THEN GOSUB 1550
1400 IF K=88 OR K=120 THEN GOSUB 1560
1410 IF K=67 OR K=99 THEN GOSUB 1650
1420 LOCATE C,L:PRINT D$
1430 IF D=0 THEN LOCATE C,L:PRINT CHR$(1
44)
1440 T(L-1,C-14)=D
1450 K$="":K=0:GOTO 1310
1460 IF L=2 THEN RETURN
1470 L=L-1:RETURN
1480 IF L=9 THEN RETURN
1490 L=L+1:RETURN
1500 IF C=15 THEN RETURN
1510 C=C-1:RETURN
1520 IF C=22 THEN RETURN
1530 C=C+1:RETURN
1540 D=1:RETURN
1550 D=0:RETURN
1560 FOR A=1 TO 8
1570 XL=0
1580 FOR B=1 TO 8
1590 XL=XL+2^(8-B)*T(A,B)
1600 S(A)=XL
1610 NEXT B
1620 LOCATE 24,A+1:PRINT XL;" "
1630 NEXT A
1640 RETURN
1650 GOSUB 1560
1660 PRINT #8
1670 PRINT #8
1680 PRINT #8
1690 FOR A=1 TO 8
1700 P$=""
1710 FOR B=1 TO 8
1720 IF T(A,B)=1 THEN P$=P$+"0"
1730 IF T(A,B)=0 THEN P$=P$+"-"
1740 NEXT B

```

```
1750 PRINT #8,P#;" ";S(A)
1760 NEXT A
1770 PRINT #8
1780 RETURN
```


DISEGNATORE GRAFICO

Con i molti caratteri grafici dell'Amstrad ed il programma **Disegnatore Grafico** si possono realizzare facilmente dei disegni a bassa risoluzione.

Il programma chiede all'inizio di inserire il colore desiderato, tra i 27 disponibili (da 0 a 26), per il bordo, il fondo e i caratteri da disegnare.

Quindi il disegno può essere realizzato tramite i quattro tasti di cursore.

Il carattere o il simbolo grafico usato in quel momento può essere cambiato premendo uno dei tasti (per i caratteri) o usando i tasti TAB e ENTER (per i simboli), tasti che fanno rispettivamente diminuire o avanzare di uno il codice del carattere visualizzato (per il set di caratteri vedere il manuale del computer).

Per cancellare basta premere lo spazio e muoversi lungo la parte da eliminare.

Infine se si preme il tasto CLR si fa ripartire il programma dall'inizio.

Il disegnatore usa un'area dello schermo di 40×24 caratteri.

```
1000 /      ** DISEGNATORE GRAFICO **
1010 / -----
1020 /           di Gaetano Marano
1030 / -----
1040 /      (c) Gruppo Editoriale Jackson
1050 / -----
1060 MODE 1:INK 1,26:INK 3,15
1070 INK 2,11:INK 0,0
1080 PEN 1:PAPER 0:BORDER 0:CLS
```

```

1090 SOUND 1,0,0,0:SOUND 2,0,0,0
1100 SOUND 4,0,0,0
1110 / -----
1120 INPUT "Inserire colore BORDO (0--26
)
     e premere RETURN ";B
1130 PRINT
1140 IF B<0 OR B>26 THEN 1120
1150 INPUT "Inserire colore FONDO (0--26
)
     e premere RETURN ";P
1160 PRINT
1170 IF P<0 OR P>26 THEN 1150
1180 INPUT "Inserire colore PENNA (0--26
)
     e premere RETURN ";I
1190 PRINT
1200 IF I<0 OR I>26 OR I=P THEN 1180
1210 INK 0,P:INK 1,I
1220 BORDER B:PAPER 0:PEN 1:CLS
1230 L=1:C=1:X=224
1240 K$=INKEY$
1250 IF K$<>" " THEN K=ASC(K$)
1260 IF K=16 THEN RUN
1270 IF K<>0 AND K<>9 AND K<>13 AND K<>1
6 AND K<>240 AND K<>241 AND K<>243 AND K
<>242 THEN X=K
1280 IF K=243 THEN GOSUB 1400
1290 IF K=242 THEN GOSUB 1380
1300 IF K=240 THEN GOSUB 1440
1310 IF K=241 THEN GOSUB 1420
1320 IF K=9 THEN GOSUB 1460
1330 IF K=13 THEN GOSUB 1490
1340 LOCATE C,L:PRINT "+"
1350 LOCATE C,L:PRINT CHR$(X)
1360 K=X
1370 GOTO 1240
1380 IF C=1 THEN RETURN
1390 C=C-1:RETURN
1400 IF C=40 THEN RETURN
1410 C=C+1:RETURN

```

```
1420 IF L=24 THEN RETURN
1430 L=L+1:RETURN
1440 IF L=1 THEN RETURN
1450 L=L-1:RETURN
1460 IF X>126 THEN X=X-1
1470 IF X=243 THEN X=239
1480 RETURN
1490 IF X<255 THEN X=X+1
1500 IF X=240 THEN X=244
1510 RETURN
```


PLOTTER

Plotter è un altro programma per disegnare, ma ad alta risoluzione (320 × 200 pixel).

All'inizio, il programma chiede di inserire il colore del bordo, del fondo e delle tre penne tra i 27 disponibili (da 0 a 26). Il disegno può essere realizzato premendo i quattro tasti di controllo cursore, che da soli disegnano nelle direzioni indicate sui tasti mentre, premuti insieme al tasto SHIFT disegnano in una direzione spostata di 45 gradi rispetto a quella indicata sul tasto (ad esempio il tasto con la freccia a destra premuto insieme a SHIFT farà muovere il puntino che disegna nella direzione in basso a destra); provare i tasti in pratica.

Per selezionare i tre colori della penna basta premere uno dei tasti da 0 a 3, in particolare il tasto 0 seleziona il colore del fondo e serve per spostarsi senza disegnare o per modificare il disegno già fatto; per ritornare a disegnare basta selezionare uno degli altri tre colori.

Infine per cancellare tutto e far ripartire il programma si può premere il tasto CLR.

```
1000 /          *** PLOTTER ***
1010 / -----
1020 /          di Gaetano Marano
1030 / -----
1040 /    (c) Gruppo Editoriale Jackson
1050 / -----
1060 MODE 1:INK 0,0:INK 1,26
1070 INK 2,11:INK 3,15
1080 PEN 1:PAPER 0:BORDER 0:CLS
1090 SOUND 1,0,0,0:SOUND 2,0,0,0
```

```

1100 SOUND 4,0,0,0
1110 / -----
1120 INPUT "Inserire colore BORDO (0--26
)          e premere RETURN ";B
1130 IF B<0 OR B>26 THEN GOTO 1120
1140 INPUT "Inserire colore FONDO (0--26
)          e premere RETURN ";P
1150 IF P<0 OR P>26 THEN GOTO 1140
1160 INPUT "Inserire colore PENNA 1 (0
--26)      e premere RETURN ";IA
1170 IF IA=P OR IA<0 OR IA>26 THEN 1160
1180 INPUT "Inserire colore PENNA 2 (0
--26)      e premere RETURN ";IB
1190 IF IB=P OR IB<0 OR IB>26 THEN 1180
1200 INPUT "Inserire colore PENNA 3 (0
--26)      e premere RETURN ";IC
1210 IF IC=P OR IC<0 OR IC>26 THEN 1200
1220 X=50:Y=50:D=1:C=1
1230 INK 1,IA:INK 2,IB:INK 3,IC
1240 BORDER B:PAPER P:CLS
1250 K$=INKEY$
1260 IF K$<>" " THEN K=ASC(K$)
1270 IF K=16 THEN RUN
1280 IF K=240 THEN GOSUB 1400
1290 IF K=243 THEN GOSUB 1420
1300 IF K=241 THEN GOSUB 1440
1310 IF K=242 THEN GOSUB 1460
1320 IF K=244 THEN GOSUB 1480
1330 IF K=247 THEN GOSUB 1500
1340 IF K=245 THEN GOSUB 1520
1350 IF K=246 THEN GOSUB 1540
1360 IF K>47 AND K<52 THEN C=K-48
1370 IF C=0 THEN PLOT X,Y,1
1380 PLOT X,Y,C
1390 K=0:GOTO 1250
1400 IF Y=398 THEN RETURN
1410 Y=Y+2:RETURN
1420 IF X=638 THEN RETURN

```

```
1430 X=X+2:RETURN
1440 IF Y=0 THEN RETURN
1450 Y=Y-2:RETURN
1460 IF X=0 THEN RETURN
1470 X=X-2:RETURN
1480 IF Y=398 OR X=638 THEN RETURN
1490 Y=Y+2:X=X+2:RETURN
1500 IF Y=0 OR X=638 THEN RETURN
1510 Y=Y-2:X=X+2:RETURN
1520 IF Y=0 OR X=0 THEN RETURN
1530 Y=Y-2:X=X-2:RETURN
1540 IF Y=398 OR X=0 THEN RETURN
1550 Y=Y+2:X=X-2:RETURN
```


CURVE MATEMATICHE

Questo programma disegna in alta risoluzione le curve di alcune **funzioni matematiche**.

```
1000 /      *** CURVE MATEMATICHE ***
1010 / -----
1020 /      di Gaetano Marano
1030 / -----
1040 /      (c) Gruppo Editoriale Jackson
1050 / -----
1060 MODE 1:INK 0,0:INK 1,26
1070 INK 2,11:INK 3,15
1080 PEN 1:PAPER 0:BORDER 1:CLS
1090 SOUND 1,0,0,0:SOUND 2,0,0,0
1100 SOUND 4,0,0,0
1110 / -----
1120 RANDOMIZE TIME
1130 FOR X=0 TO 639
1140 PLOT X,0+(X*X/1015)
1150 NEXT X
1160 FOR X=0 TO 639
1170 PLOT X,0+(X*X/1015):DRAW X,0
1180 NEXT X
1190 FOR W=1 TO 4000:NEXT W
1200 CLS
1210 INK 1,18
1220 FOR X=0 TO 639
1230 Y=RND(1)*X/1.4
1240 PLOT X,0+Y:DRAW X,0
1250 NEXT X
1260 FOR W=1 TO 4000:NEXT W
```

```
1270 CLS
1280 INK 1,15
1290 FOR X=0 TO 639
1300 Y=0+LOG(X/5+1)*82
1310 PLOT X,Y
1320 NEXT X
1330 FOR X=0 TO 639
1340 Y=0+LOG(X/5+1)*82
1350 PLOT X,Y:DRAW X,0
1360 NEXT X
1370 FOR W=1 TO 4000:NEXT W
1380 CLS
1390 INK 1,24
1400 FOR X=0 TO 639
1410 PLOT X,0+(199+COS(PI/180*(180+X*0.56)))*199)
1420 NEXT X
1430 FOR X=0 TO 639
1440 Y=199+(COS(PI/180*(180+X*0.56)))*199
)
1450 PLOT X,0+Y:DRAW X,0
1460 NEXT X
1470 FOR W=1 TO 6000:NEXT W
1480 INK 1,26:CLS
```

NEW YORK

Il programma **New York** disegna sullo schermo il panorama di una città con due file di grattacieli.

Ogni volta che si fa girare, il programma crea un panorama diverso e casuale.

```
1000 /          *** NEW YORK ***
1010 / -----
1020 /          di Gaetano Marano
1030 / -----
1040 /    (c) Gruppo Editoriale Jackson
1050 / -----
1060 MODE 0
1070 INK 0,1:INK 1,24:INK 2,20:INK 3,6
1080 INK 4,26:INK 5,0:INK 6,2:INK 7,8
1090 INK 8,10:INK 9,12:INK 10,14
1100 INK 11,16:INK 12,18:INK 13,22
1110 INK 14,1,24:INK 15,16,11
1120 PEN 1:PAPER 5:BORDER 0:QLS
1130 SOUND 1,0,0,0:SOUND 2,0,0,0
1140 SOUND 4,0,0,0
1150 / -----
1160 RANDOMIZE TIME
1170 FOR A=1 TO 30
1180 PLOT RND(1)*639,220+RND(1)*180,(8+I
NT(RND(1)*8))
1190 NEXT A
1200 K=5
1210 DIM L(15)
1220 DIM H(15)
1230 FOR A=1 TO 15
```

```

1240 L(A)=((2+INT (RND(1)*2))*20)
1250 H(A)=(K+INT (RND(1)*(2+K)))*15+INT
((RND(1)*K)-1)
1260 NEXT A
1270 X=0
1280 FOR A=1 TO 15
1290 X=X+L(A)
1300 IF X<640 THEN GOSUB 1400
1310 IF X>639 THEN GOSUB 1330
1320 NEXT A
1330 L(A)=639-(X-L(A))
1340 X=639
1350 GOSUB 1400
1360 IF K=2 THEN 1390
1370 K=2
1380 GOTO 1230
1390 GOTO 1390
1400 CC=2+INT(RND(1)*14)
1410 IF CC=14 THEN CC=0
1420 IF CC=15 THEN CC=7
1430 IF CC=5 THEN CC=12
1440 FOR Y=0 TO H(A) STEP 4
1450 PLOT X-L(A),2+Y:DRAW X-1,2+Y,CC
1460 PLOT X-L(A),0+Y:DRAW X-1,0+Y,5
1470 NEXT Y
1480 PLOT X-L(A),2:DRAW X-L(A),H(A),CC
1490 PLOT X-1,2:DRAW X-1,H(A),CC
1500 RETURN

```

DISEGNI DI LINEE

Il programma disegna sullo schermo in alta risoluzione quattro diverse **immagini grafiche** una di seguito all'altra. In particolare il terzo disegno può anche essere variato cambiando il valore della variabile V alla linea 1380 con un valore diverso, sia maggiore che minore di 12.

```
1000 /      ***  DISEGNI DI LINEE  ***
1010 / -----
1020 /           di Gaetano Marano
1030 / -----
1040 /      (c)  Gruppo Editoriale Jackson
1050 / -----
1060 MODE 1:INK 0,0:INK 1,26
1070 INK 2,11:INK 3,15
1080 PEN 1:PAPER 0:BORDER 1:CLS
1090 SOUND 1,0,0,0:SOUND 2,0,0,0
1100 SOUND 4,0,0,0
1110 / -----
1120 RANDOMIZE TIME
1130 INK 1,6
1140 FOR A=0 TO 360 STEP 2
1150 K=15+25*RND(1)
1160 RD=PI/180*A
1170 h=100+110*RND(1)
1180 PLOT 317+K*COS(RD),199+K*SIN(RD):DR
1190 AW 317+H*COS(RD),199+H*SIN(RD)
1190 NEXT A
1200 FOR W=1 TO 6000:NEXT W
1210 CLS
1220 INK 1,24
```

```

1230 FOR A=20 TO 200 STEP 8
1240 PLOT 317,A+199:DRAW 100+A,199
1250 NEXT A
1260 FOR A=20 TO 200 STEP 8
1270 PLOT 318,A+199:DRAW 535-A,199
1280 NEXT A
1290 FOR A=2 TO 180 STEP 8
1300 PLOT 317,A+4:DRAW 298-A,198
1310 NEXT A
1320 FOR A=2 TO 180 STEP 8
1330 PLOT 318,A+4:DRAW 336+A,196
1340 NEXT A
1350 FOR W=1 TO 6000:NEXT W
1360 CLS
1370 INK 1,18
1380 DEFINT X,Y:V=12
1390 DIM X(V)
1400 DIM Y(V)
1410 P=PI*2
1420 FOR N=1 TO V
1430 A=P/V*N
1440 X(N)=COS(A)*205+317
1450 Y(N)=SIN(A)*195+199
1460 NEXT N
1470 FOR A=1 TO V
1480 FOR B=1 TO V
1490 PLOT X(A),Y(A):DRAW X(B),Y(B)
1500 NEXT B
1510 NEXT A
1520 FOR W=1 TO 6000:NEXT W
1530 CLS:INK 1,15
1540 P=PI/2
1550 I=200:A=0:B=A+I:N=1
1560 R=A*P/180
1570 XA=195*COS(R)+317
1580 YA=195*SIN(R)+200
1590 R=B*P/180
1600 XB=195*COS(R)+317

```

```
1610 YB=195*SIN(R)+200
1620 PLOT XA,YA: DRAW XB,YB
1630 A=A+I: B=A+I: N=N+1
1640 IF N=19 THEN 1660
1650 GOTO 1560
1660 FOR W=1 TO 10000: NEXT W
1670 CLS: INK 1,26
```


ITALIA

La serie di codici numerici inseriti nelle istruzioni **DATA** di questo programma, contengono le coordinate per disegnare i contorni dell'Italia.

Così com'è, il programma è più che altro dimostrativo, tuttavia la parte principale che disegna l'Italia può essere inserita in altri programmi di tipo didattico, di quiz, eccetera. Il punto di inizio del disegno, dipende dai valori assegnati alle variabili X e Y (linee 1130 e 1140).

```
1000 /          *** ITALIA ***
1010 / -----
1020 /          di Gaetano Marano
1030 / -----
1040 / (c) Gruppo Editoriale Jackson
1050 / -----
1060 MODE 1
1070 INK 1,18:INK 2,6
1080 INK 3,26:INK 0,0
1090 PEN 3:PAPER 0:BORDER 1:CLS
1100 SOUND 1,0,0,0:SOUND 2,0,0,0
1110 SOUND 4,0,0,0
1120 / -----
1130 X=190:/ 0__448
1140 Y=320:/ 0__168
1150 RESTORE
1160 K=0:C=1
1170 READ D:K=K+1
1180 IF D=1111 THEN 1260
1190 D=D*2
1200 IF D=0 THEN Y=Y-2:GOTO 1250
```

```
1210 IF K>220 THEN C=3
1220 IF K>375 THEN C=2
1230 D=D-52
1240 PLOT X+D,Y,C
1250 GOTO 1170
1260 GOTO 1260
1270 DATA 94,0,88,89,91,92,93,0,86,87,88
,90,93,0,82,83,85,94,0,81,83,84,95,0
1280 DATA 79,82,96,97,98,99,0,78,80,81,1
00,101,102,103,0,64,65,72,78,104,105,0,
1290 DATA 64,65,71,73,74,75,76,77,78,103
,0,64,66,71,103,104,0
1300 DATA 63,67,68,70,105,0,59,61,62,68,
70,104,0,54,55,56,57,58,60,69,104,0
1310 DATA 53,101,104,0,54,100,101,102,10
3,104,105,0,55,100,101,106,107,0
1320 DATA 55,99,107,0,56,95,96,97,98,0,5
5,94,0,54,94,0
1330 DATA 52,53,94,0,52,95,0,53,95,0,53,
96,0,54,94,95,96,0,53,94,0
1340 DATA 53,94,0,53,94,0,53,67,68,69,94
,0,53,65,66,70,71,94,0
1350 DATA 54,55,64,72,73,95,0,56,57,58,5
9,63,73,74,95,0,59,62,75,76,96,0
1360 DATA 59,62,77,97,0,58,59,60,61,78,9
8,0,78,99,100,0,79,101,0,79,102,0
1370 DATA 79,103,0,79,104,105,0,80,105,0
,80,105,0,81,106,0,81,106,0,75,81,107,0
1380 DATA 81,82,107,0,83,107,0,77,78,79,
83,107,0,79,84,108,0,76,85,108,0
1390 DATA 86,109,0,86,87,88,110,0,83,85,
89,111,0,90,112,0,90,113,114,0
1400 DATA 91,92,115,0,93,116,117,0,94,11
8,119,122,123,124,125,0,94,120,121,126,0
1410 DATA 95,125,0,96,124,0,97,124,0,98,
99,125,126,0,69,100,102,103,127,128,129
1420 DATA 0,62,69,101,104,105,106,107,13
0,0,61,67,68,70,71,108,131,132,133,0
```

1430 DATA 66,72,108,134,135,0,60,65,71,1
09,136,0
1440 DATA 60,61,62,63,64,72,109,110,111,
137,138,0,60,72,109,112,139,140,0
1450 DATA 60,61,72,107,113,114,115,141,1
42,0,62,73,112,113,116,134,135,143,0
1460 DATA 62,73,110,117,133,136,144,0,62
,72,117,132,137,138,139,140,145,0
1470 DATA 62,72,117,132,142,146,0,63,71,
117,118,131,143,147,0
1480 DATA 62,72,119,122,131,143,146,0,62
,72,120,121,123,130,143,144,146,0
1490 DATA 62,72,123,124,131,145,146,0,63
,72,124,130,0,63,72,124,130,0
1500 DATA 62,63,72,125,130,0,62,71,125,1
31,132,133,0,61,71,126,134,0
1510 DATA 62,71,126,135,0,61,71,127,135,
0,59,61,67,68,69,71,127,135,0
1520 DATA 62,66,70,127,135,0,61,63,66,12
7,135,0,63,64,65,128,133,134,135,0
1530 DATA 128,132,0,128,131,0,125,126,12
7,131,0,118,126,131,0,116,126,131,0
1540 DATA 126,130,131,0,118,125,129,0,12
2,124,125,128,0
1550 DATA 103,117,120,121,122,124,128,0,
98,101,102,103,114,115,116,118,119
1560 DATA 122,124,128,0,92,97,99,100,104
,105,109,110,111,112,113,121,124,125
1570 DATA 126,127,0,94,96,106,107,108,12
0,0,96,120,0,96,119,0,97,119,0
1580 DATA 98,99,100,101,119,0,102,103,11
8,0,104,118,0,105,106,118,0
1590 DATA 107,108,120,0,109,110,111,112,
121,0,113,120,0,114,119,0
1600 DATA 91,114,115,118,119,0,92,116,11
7,0,1111

NUMERI DIGITALI

Se si vogliono inserire dei numeri di varie dimensioni nel modo grafico ad alta risoluzione, si può usare la routine **Numeri Digitali** che consente appunto di scrivere delle cifre digitali a sette segmenti.

Per scrivere un numero bisogna inserire il numero stesso (tra 0 e 9) nella variabile N, la larghezza dei segmenti orizzontali (in pixel) nella variabile L, l'altezza dei segmenti verticali nella variabile H e le coordinate alle quali va scritto il numero nelle variabili X e Y.

Se le cifre devono essere tutte della stessa dimensione, i valori L e H possono essere stabiliti una sola volta all'inizio, inoltre se si devono scrivere più numeri di seguito, occorre ad ogni numero incrementare il valore di X per non sovrapporre le cifre.

```
100 /          PROGRAMMA DIMOSTRATIVO
102 / -----
104 MODE 1:INK 0,0:INK 1,26
106 INK 2,11:INK 3,15
108 PEN 1:PAPER 0:BORDER 1:CLS
110 SOUND 1,0,0,0:SOUND 2,0,0,0
112 SOUND 4,0,0,0
114 RANDOMIZE TIME
116 FOR A=1 TO 20
118 X=20+RND(1)*580
120 Y=100+RND(1)*280
122 L=10+RND(1)*30
124 H=10+RND(1)*30
126 N=RND(1)*10
```

```

128 C=1+INT(RND(1)*3)
130 GOSUB 7000
132 NEXT A
134 FOR W=1 TO 8000:NEXT W:CLS:END
136 / -----
7000 /          ** NUMERI DIGITALI **
7010 / -----
7020 /          di Gaetano Marano
7030 / -----
7040 / (c) Gruppo Editoriale Jackson
7050 / -----
7060 N=INT(N):IF N<0 OR N>9 THEN RETURN
7070 IF N=0 OR N=2 OR N=3 OR N=5 OR N=6
OR N=7 OR N=8 OR N=9 THEN PLOT X,Y-4:DR
W X+L,Y-4,C
7080 IF N=2 OR N=3 OR N=4 OR N=5 OR N=6
OR N=8 OR N=9 THEN PLOT X,Y-4-H:DRAW X+L
,Y-4-H,C
7090 IF N=0 OR N=2 OR N=3 OR N=5 OR N=6
OR N=8 OR N=9 THEN PLOT X,Y-4-H-H:DRAW X
+L,Y-4-H-H,C
7100 IF N=0 OR N=1 OR N=2 OR N=3 OR N=4
OR N=7 OR N=8 OR N=9 THEN PLOT X+L,Y-4:D
RAW X+L,Y-4-H,C
7110 IF N=0 OR N=1 OR N=3 OR N=4 OR N=5
OR N=6 OR N=7 OR N=8 OR N=9 THEN PLOT X+
L,Y-4-H:DRAW X+L,Y-4-H-H,C
7120 IF N=0 OR N=2 OR N=6 OR N=8 THEN PL
OT X,Y-4-H:DRAW X,Y-4-H-H,C
7130 IF N=0 OR N=4 OR N=5 OR N=6 OR N=8
OR N=9 THEN PLOT X,Y-4:DRAW X,Y-4-H,C
7140 RETURN

```

LETTURA VELOCE

Questo programma può essere usato per incrementare la propria **velocità di lettura**. In una serie di istruzioni DATA sono contenute 200 parole che vengono visualizzate in maniera casuale e per breve tempo in un riquadro sullo schermo. La prima frase viene sempre visualizzata per un tempo di 0.1 secondi che viene indicato in anticipo.

Sparita la frase il computer chiede di scriverla, quindi confronta la frase scritta con quella appena visualizzata.

Se la frase inserita è uguale a quella scelta dal computer esce la scritta '**Frase: * ESATTA ***', se invece la frase inserita è diversa da quella visualizzata esce la scritta '**Frase: SBAGLIATA!!!**'.

Inoltre, ogni volta che si riconosce esattamente una frase, il tempo di visualizzazione della successiva viene dimezzato (fino ad un minimo di 0.02 secondi), mentre ogni volta che si sbaglia viene raddoppiato (fino ad un massimo di 1 sec.).

Alle 200 parole già incluse nel programma se ne possono **aggiungere** molte altre, inserendo le nuove parole o frasi (di massimo 20 caratteri) in una serie di istruzioni DATA; l'ultima linea deve essere sempre uguale alla linea 9998 (vedere listato) che contiene un carattere usato dal programma per riconoscere la fine della lista durante il conteggio delle frasi.

```
1000 /      *** LETTURA VELOCE ***
1010 / -----
1020 /      di Gaetano Marano
1030 / -----
1040 /      (c) Gruppo Editoriale Jackson
1050 / -----
```

```

1060 MODE 1:INK 1,26:INK 0,0
1070 INK 2,11:INK 3,15
1080 PEN 2:PAPER 0:BORDER 0:CLS
1090 SOUND 1,0,0,0:SOUND 2,0,0,0
1100 SOUND 4,0,0,0
1110 / -----
1120 A$=CHR$(150):D$=CHR$(154):DD$=CHR$(
154):E$=CHR$(156):F$=CHR$(149):G$=CHR$(1
47):H$=CHR$(153)
1130 FOR W=1 TO 4
1140 D$=D$+D$
1150 NEXT W
1160 DD$=DD$+DD$
1170 D$=D$+DD$+DD$
1180 B$=" "
1190 B$=B$+B$
1200 LOCATE 10,5
1210 PRINT A$;D$;E$
1220 LOCATE 10,6
1230 PRINT F$;" "
1240 LOCATE 10,7
1250 PRINT G$;D$;H$
1260 PEN 1
1270 RESTORE
1280 FOR A=1 TO 10000
1290 READ C$:IF C$="#" THEN 1310
1300 NEXT A
1310 NF=A-1
1320 DIM T$(6)
1330 T$(1)="0180.02 Sec."
1340 T$(2)="0450.05 Sec."
1350 T$(3)="0910.1 Sec. "
1360 T$(4)="1820.2 Sec. "
1370 T$(5)="4550.5 Sec. "
1380 T$(6)="909 1 Sec. "
1390 TX=3
1400 PEN 1:FOR T=1 TO 2000:NEXT T
1410 RANDOMIZE TIME

```



```

1420 LOCATE 10,10:PEN 3:PRINT "Durata fr
ase "; RIGHT$(T$(TX),9)
1430 FOR T=1 TO 1000:NEXT T
1440 PEN 1
1450 RESTORE
1460 FOR A=1 TO INT(NF*RND(1))
1470 READ F$:NEXT A
1480 P=VAL (LEFT$ (T$(TX),3))
1490 READ F$:IF LEN (F$)>20 THEN F$=LEFT
$(F$,20)
1500 LOCATE 11,6:PRINT F$
1510 FOR T=1 TO P:NEXT T
1520 LOCATE 11,6:PRINT "
"
1530 FOR T=1 TO 750:NEXT T
1540 LOCATE 3,13
1550 PRINT " Scrivete la frase che ave
te visto e premete RETURN "
1560 LOCATE 4,16
1570 INPUT X$
1580 R=0:IF UPPER$(X$)=UPPER$(F$) THEN R
=1
1590 IF R=1 THEN LOCATE 10,10:PEN 1:PRIN
T " Frase: * ESATTA * ";B$;B$;B$;B$;
B$;B$
1600 IF R=0 THEN LOCATE 10,10:PEN 1:PRIN
T " Frase: SBAGLIATA !! ";B$;B$;B$;B$
;B$;B$
1610 IF R=1 AND TX>1 THEN TX=TX-1
1620 IF R=0 AND TX<6 THEN TX=TX+1
1630 GOTO 1400
1640 /-----/
2000 / CATALOGO PAROLE E FRASI
2001 DATA "BIANCO"
2002 DATA "ROSSO"
2003 DATA "VERDE"
2004 DATA "GIALLO"
2005 DATA "BLU"

```

2006 DATA "NERO"
2007 DATA "ROMA"
2008 DATA "MILANO"
2009 DATA "FIRENZE"
2010 DATA "NAPOLI"
2011 DATA "PARIGI"
2012 DATA "BERLINO"
2013 DATA "LONDRA"
2014 DATA "MADRID"
2015 DATA "NEW YORK"
2016 DATA "LOS ANGELES"
2017 DATA "ATENE"
2018 DATA "TOKYO"
2019 DATA "MOSCA"
2020 DATA "BUENOS AIRES"
2021 DATA "GINEVRA"
2022 DATA "MICROPROCESSORE"
2023 DATA "KEYBOARD"
2024 DATA "MONITOR"
2025 DATA "AMSTRAD"
2026 DATA "AEREO"
2027 DATA "NAVE"
2028 DATA "TREN0"
2029 DATA "COMPUTER"
2030 DATA "FOTOGRAFIA"
2031 DATA "ATTRICE"
2032 DATA "AUTOMOBILE"
2033 DATA "CREDIT CARD"
2034 DATA "LINEA"
2035 DATA "ARTICOLO"
2036 DATA "INDIRIZZO"
2037 DATA "MOTORE"
2038 DATA "CAPIRE"
2039 DATA "VOLARE"
2040 DATA "CAMMINARE"
2041 DATA "NAPOLEONE"
2042 DATA "GARIBALDI"
2043 DATA "CESARE"

2044 DATA "WASHINGTON"
2045 DATA "PRINTER"
2046 DATA "UOMO"
2047 DATA "DONNA"
2048 DATA "CASA"
2049 DATA "SCALA"
2050 DATA "ASCENSORE"
2051 DATA "RUOTA"
2052 DATA "LIBRO"
2053 DATA "JACKSON"
2054 DATA "BYTE"
2055 DATA "BUONO"
2056 DATA "CATTIVO"
2057 DATA "VELOCE"
2058 DATA "LENTO"
2059 DATA "GRANDE"
2060 DATA "PICCOLO"
2061 DATA "LEGGERO"
2062 DATA "PESANTE"
2063 DATA "MARIO"
2064 DATA "LAURA"
2065 DATA "FRANCO"
2066 DATA "ANTONIO"
2067 DATA "MARIA"
2068 DATA "FRANCESCO"
2069 DATA "BARBARA"
2070 DATA "GATTI"
2071 DATA "ELEFANTE"
2072 DATA "DELFINO"
2073 DATA "GIRAFFA"
2074 DATA "ANTILOPE"
2075 DATA "LEONE"
2076 DATA "CONDOR"
2077 DATA "TIGRE"
2078 DATA "DOBERMAN"
2079 DATA "IPPOPOTAMO"
2080 DATA "CANE"
2081 DATA "BASIC"

2082 DATA "PASCAL"
2083 DATA "FORTRAN"
2084 DATA "FORTH"
2085 DATA "LISP"
2086 DATA "PROLOG"
2087 DATA "LOGO"
2088 DATA "COBOL"
2089 DATA "APL"
2090 DATA "CALENDARIO"
2091 DATA "TRANSISTOR"
2092 DATA "STOFFA"
2093 DATA "TEATRO"
2094 DATA "RIVISTA"
2095 DATA "SEMAFORO"
2096 DATA "VOLERE"
2097 DATA "CANTARE"
2098 DATA "JET"
2099 DATA "SCRIVERE"
2100 DATA "LEGGERE"
2101 DATA "CALIFORNIA"
2102 DATA "AMICO"
2103 DATA "SAPORI"
2104 DATA "PROGRAMMA"
2105 DATA "DEBUG"
2106 DATA "DISK"
2107 DATA "MUSICA"
2108 DATA "CANTANTE"
2109 DATA "TELEFONO"
2110 DATA "TELEGRAFO"
2111 DATA "TELEX"
2112 DATA "CASSETTA"
2113 DATA "LOCALI"
2114 DATA "NETWORK"
2115 DATA "JAZZ"
2116 DATA "STATI"
2117 DATA "FORMULA"
2118 DATA "BORSA"
2119 DATA "PARTICELLA"

2120 DATA "PROTONE"
2121 DATA "QUARK"
2122 DATA "NEUTRINO"
2123 DATA "ELETTRONE"
2124 DATA "ACQUA"
2125 DATA "ARANCIA"
2126 DATA "ANANAS"
2127 DATA "POMODORO"
2128 DATA "KIWI"
2129 DATA "TESTA"
2130 DATA "BRACCIO"
2131 DATA "GAMBA"
2132 DATA "OCCHI"
2133 DATA "NASO"
2134 DATA "CUORE"
2135 DATA "UDIRE"
2136 DATA "VEDERE"
2137 DATA "GIACCA"
2138 DATA "CRAVATTA"
2139 DATA "METRO"
2140 DATA "INCH"
2141 DATA "CACCIAVITE"
2142 DATA "CARBURATORE"
2143 DATA "STETOSCOPIO"
2144 DATA "PAROLE"
2145 DATA "PAGINE"
2146 DATA "CONTENITORE"
2147 DATA "DECOLLO"
2148 DATA "DESIGN"
2149 DATA "ARTISTA"
2150 DATA "RAGIONE"
2151 DATA "NASTRO"
2152 DATA "MICROFONO"
2153 DATA "STRUMENTO"
2154 DATA "PIANOFORTE"
2155 DATA "CLARINETTO"
2156 DATA "SAX"
2157 DATA "SYNTHESIZER"

2158 DATA "ECONOMIA"
2159 DATA "MONETA"
2160 DATA "DOLLARO"
2161 DATA "STERLINA"
2162 DATA "CAMBIO"
2163 DATA "SOFFITTA"
2164 DATA "PANINO"
2165 DATA "SPAGHETTI"
2166 DATA "HAMBURGER"
2167 DATA "WHISKY"
2168 DATA "CANCELLARE"
2169 DATA "MEMORIA"
2170 DATA "AUTOBUS"
2171 DATA "COMBINAZIONE"
2172 DATA "URAGANO"
2173 DATA "SPACE SHUTTLE"
2174 DATA "ASTRONAUTA"
2175 DATA "PIOGGIA"
2176 DATA "CAMINETTO"
2177 DATA "STRADA"
2178 DATA "PERCORRERE"
2179 DATA "INDUSTRIA"
2180 DATA "INGRANAGGIO"
2181 DATA "RIVISTA"
2182 DATA "UNIX"
2183 DATA "BLOCCHETTO"
2184 DATA "DINOSAURO"
2185 DATA "DIZIONARIO"
2186 DATA "VIDEOGAMES"
2187 DATA "STAR WARS"
2188 DATA "PRODUZIONE"
2189 DATA "ALLENAMENTO"
2190 DATA "BASKET"
2191 DATA "DISPLAY"
2192 DATA "PIXEL"
2193 DATA "LOGARITMO"
2194 DATA "TRIGONOMETRIA"
2195 DATA "HOBBY"

```
2196 DATA "TELEFILM"  
2197 DATA "NOTIZIE"  
2198 DATA "SOPRANO"  
2199 DATA "PENULTIMA"  
2200 DATA "ULTIMA"  
9998 DATA "#"  
9999 / FINE LISTA
```


CRONOMETRO

Il programma **Cronometro** permette di misurare i tempi e di visualizzarli in ore, minuti e secondi.

Il cronometro è controllato dai tasti **Z** (Start), **X** (Lap, tempo intermedio) e **C** (Stop).

```
1000 /      *** CRONOMETRO ***
1010 / -----
1020 /      di Gaetano Marano
1030 / -----
1040 / (c) Gruppo Editoriale Jackson
1050 / -----
1060 /      funzione          tasto
1070 / -----
1080 /      START              Z
1090 /      LAP                X
1100 /      STOP              C
1110 / -----
1120 MODE 1:INK 0,1:INK 1,26
1130 INK 2,24:INK 3,15
1140 PEN 2:PAPER 0:BORDER 1:CLS.
1150 SOUND 1,0,0,0:SOUND 2,0,0,0
1160 SOUND 4,0,0,0
1170 / -----
1180 LOCATE 2,3:PRINT "  |||||
      |||||"
1190 LOCATE 2,4:PRINT "  |||||
      |||||"
1200 LOCATE 2,5:PRINT "  |||||
      |||||"
```

```

1210 LOCATE 14,2:PRINT "hr min sec"
1220 LOCATE 4,8:PRINT "tasti: Z_START
X_LAP C_STOP"
1230 LOCATE 14,4:PRINT "00 : 00 : 00"
1240 K$=INKEY$
1250 IF K$<>" " THEN K=ASC(K$)
1260 IF K=90 OR K=122 THEN GOTO 1290
1270 IF K=88 OR K=120 THEN K=5:GOTO 1310
1280 GOTO 1240
1290 V=TIME
1300 LOCATE 14,4:PRINT "00 : 00 : 00"
1310 GOSUB 1370
1320 K$=INKEY$
1330 IF K$<>" " THEN K=ASC(K$)
1340 IF K=67 OR K=99 THEN GOTO 1240
1350 IF K=88 OR K=120 THEN K=5:GOTO 1240
1360 GOTO 1310
1370 T=INT((TIME-V)/30)
1380 H=INT(T/36000)
1390 T=T-H*36000
1400 M=INT(T/600)
1410 T=T-M*600
1420 S=INT(T/10)
1430 C=T-S*10
1440 H$=STR$(H)
1450 M$=STR$(M)
1460 S$=STR$(S)
1470 C$=STR$(C)
1480 LOCATE 16-LEN(H$),4:PRINT H$
1490 LOCATE 21-LEN(M$),4:PRINT M$
1500 LOCATE 26-LEN(S$),4:PRINT S$
1510 LOCATE 27,4:PRINT C$
1520 RETURN

```

SCRITTE PUBBLICITARIE

Il programma chiede di inserire per prima cosa un testo di massimo 200 caratteri che può contenere, ad esempio, un **messaggio pubblicitario**, una serie di prezzi, delle quotazioni di borsa, delle offerte speciali, dei comunicati importanti, eccetera. Quindi si deve inserire la velocità di rotazione delle scritte con un valore tra 1 (più veloce) a 10 (più lenta). Per inserire una frase diversa occorre fermare il programma premendo due volte il tasto **ESC** e quindi dare di nuovo **RUN**.

```
1000 /      ** SCRITTE PUBBLICITARIE **
1010 / -----
1020 /      di Gaetano Marano
1030 / -----
1040 /      (c) Gruppo Editoriale Jackson
1050 / -----
1060 MODE 1:INK 0,1:INK 1,24
1070 INK 2,26:INK 3,15
1080 PEN 1:PAPER 0:BORDER 1:CLS
1090 SOUND 1,0,0,0:SOUND 2,0,0,0
1100 SOUND 4,0,0,0
1110 / -----
1120 W#=CHR$(154)
1130 FOR Q=1 TO 39
1140 W#=W#+CHR$(154)
1150 NEXT Q
1160 INPUT "Inserire frase (max 200 caracteri)
e premere RETURN ";A$
1170 CLS
1180 INPUT "Inserire velocità (1-10)
e premere RETURN ";T:IF T<1
```

```

OR T>10 THEN 1180
1190 CLS
1200 PEN 3
1210 LOCATE 1,4:PRINT W$
1220 LOCATE 1,5:PRINT W$
1230 LOCATE 1,9:PRINT W$
1240 LOCATE 1,10:PRINT W$
1250 PEN 1
1260 S$=SPACE$(40)
1270 IF LEN(A$)<40 THEN H$=LEFT$(S$,40-LEN(A$))
1280 IF LEN(A$)>39 THEN H$=" "
1290 B$=A$+H$
1300 C$=B$
1310 LOCATE 1,7:PRINT LEFT$(C$,40)
1320 C$=RIGHT$(C$,LEN(C$)-1)
1330 IF LEN(C$)<40 THEN C$=C$+B$
1340 FOR D=1 TO 20*T:NEXT D
1350 GOTO 1310

```

CONTAPEZZI

Un uso insolito del computer può essere quello di **Contapezzi** capace di contare da 0 a 99.999.999 pezzi.

Facendo girare il programma, ogni volta che si preme RETURN il numero viene incrementato di uno. Oltre a ciò, si può inserire anche un numero qualsiasi da aggiungere (per esempio 334925) che premendo RETURN viene sommato al numero presente sullo schermo oppure sottratto, se il numero inserito è preceduto dal segno —.

Se si cerca di sommare o di sottrarre un numero che farebbe scendere il totale sotto lo 0 o lo farebbe salire oltre il massimo, il programma fa uscire sul video la frase 'NUMERO ERATO O ECCESSIVO'.

```
1000 /          *** CONTAPEZZI ***
1010 / -----
1020 /          di Gaetano Marano
1030 / -----
1040 / (c) Gruppo Editoriale Jackson
1050 / -----
1060 MODE 1:INK 0,0:INK 1,26
1070 INK 2,11:INK 3,15
1080 PEN 1:PAPER 0:BORDER 0:CLS
1090 SOUND 1,0,0,0:SOUND 2,0,0,0
1100 SOUND 4,0,0,0
1110 / -----
1120 SYMBOL 255,96,144,144,96,0,0,0
1130 B#=CHR$(150):C#=CHR$(156):D#=CHR$(1
47):E#=CHR$(153):F#=CHR$(151):G#=CHR$(15
7)
```

```

1140 H$=CHR$(158):I$=CHR$(155):L$=CHR$(1
49):M$=CHR$(145):Q$=CHR$(255):X$=CHR$(14
8)
1150 SP$=SPACE$(40)
1160 PEN 2
1170 PRINT "    Premere RETURN per un sol
o pezzo"
1180 PRINT
1190 PRINT "    o introdurre N";Q$;" pezz
i in piu' o"
1200 PRINT
1210 PRINT "    in meno (con -) e premere
RETURN."
1220 PEN 1
1230 LOCATE 5,12:PRINT B$;C$;B$;C$;B$;C$
;B$;C$;H$
1240 LOCATE 5,13:PRINT F$;E$;F$;" ";B$;E
$;B$;E$;L$
1250 LOCATE 5,14:PRINT M$;" ";D$;E$;D$;E
$;D$;E$;I$
1260 N=0
1270 C=0
1280 PEN 3
1290 LOCATE 5,22:INPUT N$
1300 PEN 1
1310 LOCATE 1,22:PRINT SP$
1320 IF N$="" THEN N$="1"
1330 IF ASC(N$)<>45 THEN P$=N$:GOTO 1350
1340 IF ASC(N$)=45 THEN P$=RIGHT$(N$,LEN
(N$)-1):GOTO 1400
1350 FOR A=1 TO LEN(N$)
1360 IF ASC(P$)<48 OR ASC(P$)>57 THEN GO
TO 1580
1370 P$=RIGHT$(P$,LEN(P$)-1)
1380 NEXT A
1390 GOTO 1440
1400 FOR A=2 TO LEN(N$)
1410 IF ASC(P$)<48 OR ASC(P$)>57 THEN GO

```

```

TO 1580
1420 P$=RIGHT$(P$,LEN(P$)-1)
1430 NEXT A
1440 IF (N+VAL(N$))>(100000000-1) OR (N+
VAL(N$))<0 THEN GOTO 1580
1450 N=N+VAL(N$)
1460 N$="1"
1470 A$=STR$(N)
1480 A$=RIGHT$(A$,LEN(A$)-1)
1490 IF LEN(A$)<8 THEN A$=A$+LEFT$("
",8-LEN(A$))
1500 FOR P=1 TO 8
1510 C=P*2+15
1520 J=ASC(A$)-47
1530 IF J=-15 THEN J=11
1540 ON J GOSUB 1630,1670,1710,1750,1790
,1830,1870,1910,1950,1990,2030
1550 A$=RIGHT$(A$,LEN(A$)-1)
1560 NEXT P
1570 GOTO 1270
1580 LOCATE 5,18: PEN 2: PAPER 1: PRINT "
NUMERO ERRATO 0 ECCESSIVO ": PEN 1: PAPER
0
1590 A$="0"
1600 FOR T=1 TO 2000: NEXT T
1610 LOCATE 1,18: PRINT SP$
1620 GOTO 1270
1630 LOCATE C,12: PRINT B$;C$
1640 LOCATE C,13: PRINT L$;L$
1650 LOCATE C,14: PRINT D$;E$
1660 RETURN
1670 LOCATE C,12: PRINT " ";C$
1680 LOCATE C,13: PRINT " ";L$
1690 LOCATE C,14: PRINT " ";I$
1700 RETURN
1710 LOCATE C,12: PRINT B$;C$
1720 LOCATE C,13: PRINT B$;E$
1730 LOCATE C,14: PRINT D$;E$

```

```
1740 RETURN
1750 LOCATE C,12:PRINT B$;C$
1760 LOCATE C,13:PRINT " ";G$
1770 LOCATE C,14:PRINT D$;E$
1780 RETURN
1790 LOCATE C,12:PRINT X$;X$
1800 LOCATE C,13:PRINT D$;G$
1810 LOCATE C,14:PRINT " ";M$
1820 RETURN
1830 LOCATE C,12:PRINT B$;C$
1840 LOCATE C,13:PRINT D$;C$
1850 LOCATE C,14:PRINT D$;E$
1860 RETURN
1870 LOCATE C,12:PRINT B$;C$
1880 LOCATE C,13:PRINT F$;C$
1890 LOCATE C,14:PRINT D$;E$
1900 RETURN
1910 LOCATE C,12:PRINT B$;C$
1920 LOCATE C,13:PRINT " ";L$
1930 LOCATE C,14:PRINT " ";M$
1940 RETURN
1950 LOCATE C,12:PRINT B$;C$
1960 LOCATE C,13:PRINT F$;G$
1970 LOCATE C,14:PRINT D$;E$
1980 RETURN
1990 LOCATE C,12:PRINT B$;C$
2000 LOCATE C,13:PRINT D$;G$
2010 LOCATE C,14:PRINT D$;E$
2020 RETURN
2030 LOCATE C,12:PRINT " "
2040 LOCATE C,13:PRINT " "
2050 LOCATE C,14:PRINT " "
2060 RETURN
```


FISICA DEI REATTORI NUCLEARI

Il programma simula il processo di 'fissione nucleare' che è alla base del funzionamento appunto delle **centrali nucleari** che producono energia.

La fissione nucleare è una reazione provocata da neutroni veloci che colpiscono degli atomi di Uranio 235 i quali si spaccano producendo due atomi più piccoli di Bario e Krypto ed altri due neutroni che colpendo a loro volta degli atomi di Uranio provocano la cosiddetta 'reazione a catena'. La reazione produce anche energia sfruttata per produrre calore ed elettricità.

Il programma produce un'animazione che può essere vista in modo lento (premendo il tasto **L**) o in modo veloce (tasto **V**), il tasto **R** fa ritornare il programma allo stato iniziale.

```
1000 /      FISICA DEI REATTORI
1010 /      NUCLEARI
1020 / -----
1030 /      di Gaetano Marano
1040 / -----
1050 /      (c) Gruppo Editoriale Jackson
1060 / -----
1070 MODE 1:INK 0,0:INK 1,26
1080 INK 2,11:INK 3,15
1090 PEN 2:PAPER 0:BORDER 1:CLS
1100 SOUND 1,0,0,0:SOUND 2,0,0,0
1110 SOUND 4,0,0,0
1120 / -----
1130 Q$=CHR$(144):W$=CHR$(227):P$=CHR$(2
31)
```

```

1140 LOCATE 6,1:PRINT " SIMULATORE FISSI
ONE NUCLEARE"
1150 LOCATE 1,2:PRINT "-----"
-----"
1160 LOCATE 1,3:PRINT "L_Lento"
1170 LOCATE 1,5:PRINT "V_Veloce"
1180 LOCATE 1,7:PRINT "R_Ripete"
1190 LOCATE 28,3:PEN 1:PRINT Q$;"_Neutro
ne"
1200 LOCATE 28,4:PEN 2:PRINT P$;"_Uranio
235"
1210 LOCATE 28,5:PEN 3:PRINT "o_Bario 13
9":PEN 2
1220 LOCATE 28,6:PEN 1:PRINT W$;"_Krypto
97":PEN 2
1230 LOCATE 20,3:PEN 1:PRINT Q$:PEN 2
1240 LOCATE 20,7:PRINT P$
1250 LOCATE 13,14:PRINT P$
1260 LOCATE 27,14:PRINT P$
1270 LOCATE 8,19:PRINT P$
1280 LOCATE 18,19:PRINT P$
1290 LOCATE 22,19:PRINT P$
1300 LOCATE 32,19:PRINT P$
1310 K$=INKEY$
1320 IF K$="I" OR K$="L" THEN T=100:K$="
":GOTO 1350
1330 IF K$="O" OR K$="V" THEN T=1:K$="":
GOTO 1350
1340 GOTO 1310
1350 FOR A=4 TO 7
1360 PEN 1
1370 GOSUB 1790
1380 LOCATE 20,A:PRINT Q$
1390 LOCATE 20,A-1:PRINT " "
1400 NEXT A
1410 LOCATE 19,7:PEN 3:PRINT "o ":PEN 1:
LOCATE 21,7:PRINT W$:PEN 2
1420 PEN 1

```

```

1430 FOR A=9 TO 14
1440 GOSUB 1790
1450 LOCATE 26-A,A:PRINT Q$
1460 LOCATE 14+A,A:PRINT Q$
1470 LOCATE 27-A,A-1:PRINT " "
1480 LOCATE 13+A,A-1:PRINT " "
1490 NEXT A
1500 LOCATE 12,14:PEN 3:PRINT "o ":PEN 1
:LOCATE 14,14:PRINT W$
1510 LOCATE 26,14:PEN 3:PRINT "o ":PEN 1
:LOCATE 28,14:PRINT W$
1520 FOR A=16 TO 19
1530 GOSUB 1790
1540 LOCATE 40-A,A:PRINT Q$
1550 LOCATE 26-A,A:PRINT Q$
1560 LOCATE 14+A,A:PRINT Q$
1570 LOCATE A,A:PRINT Q$
1580 LOCATE 41-A,A-1:PRINT " "
1590 LOCATE 27-A,A-1:PRINT " "
1600 LOCATE 13+A,A-1:PRINT " "
1610 LOCATE -1+A,A-1:PRINT " "
1620 NEXT A
1630 LOCATE 7,19:PEN 3:PRINT "o ":LOCATE
9,19:PEN 1:PRINT W$:PEN 3:LOCATE 17,19:
PRINT "o ":PEN 1
1640 LOCATE 19,19:PRINT W$:PEN 3:LOCATE
21,19:PRINT "o ":PEN 1:LOCATE 23,19:PRIN
T W$
1650 PEN 3:LOCATE 31,19:PRINT "o ":LOCAT
E 33,19:PEN 1:PRINT W$
1660 GOSUB 1790
1670 LOCATE 7,20:PRINT Q$;" ";Q$;"
";Q$;" ";Q$;" ";Q$;" ";Q$;"
";Q$
" ";Q$
1680 GOSUB 1790
1690 LOCATE 6,20:PRINT "
"
1700 LOCATE 6,21:PRINT Q$;" ";Q$;"

```

```

";Q$;"   ";Q$;" ";Q$;"   ";Q$;"   ";Q$;"
"   ";Q$
1710 GOSUB 1790
1720 LOCATE 6,21:PRINT "
"
1730 LOCATE 5,22:PRINT Q$;"   ";Q$;"
";Q$;"   ";Q$;"   ";Q$;"   ";Q$;"
"   ";Q$
1740 GOSUB 1790
1750 LOCATE 5,22:PRINT "
"
1760 K$=INKEY$
1770 IF K$="r" OR K$="R" THEN K$="":RUN
1780 GOTO 1760
1790 FOR D=1 TO T
1800 NEXT D:RETURN

```

PIANO 4 OTTAVE 49 NOTE

Il programma trasforma un computer Amstrad in un **Piano a 4 ottave** per un totale di 49 note, che possono essere suonate tramite 25 tasti più il tasto SHIFT.

Il piano va utilizzato con la tastiera in modo minuscole (senza premere quindi il tasto CAPS LOCK), inoltre per ascoltare le note ricordarsi di alzare il volume dell'altoparlante interno tramite l'apposita manopola sul lato destro del computer. I tasti che producono le note sono indicati nella tabella in basso (il segno + indica il diesis).

tasto	nota	ottava	tasto	nota	ottava
Q	DO	1	SHIFT Q	DO	3
2	DO+	1	SHIFT 2	DO+	3
W	RE	1	SHIFT W	RE	3
3	RE+	1	SHIFT 3	RE+	3
E	MI	1	SHIFT E	MI	3
R	FA	1	SHIFT R	FA	3
5	FA+	1	SHIFT 5	FA+	3
T	SOL	1	SHIFT T	SOL	3
6	SOL+	1	SHIFT 6	SOL+	3
Y	LA	1	SHIFT Y	LA	3
7	LA+	1	SHIFT 7	LA+	3
U	SI	1	SHIFT U	SI	3
Z	DO	2	SHIFT Z	DO	4
S	DO+	2	SHIFT S	DO+	4
X	RE	2	SHIFT X	RE	4
D	RE+	2	SHIFT D	RE+	4
C	MI	2	SHIFT C	MI	4

tasto	nota	ottava	tasto	nota	ottava
V	FA	2	SHIFT V	FA	4
G	FA+	2	SHIFT G	FA+	4
B	SOL	2	SHIFT B	SOL	4
H	SOL+	2	SHIFT H	SOL+	4
N	LA	2	SHIFT N	LA	4
J	LA+	2	SHIFT J	LA+	4
M	SI	2	SHIFT M	SI	4
,	DO	3	SHIFT ,	DO	5

```

1000 /  ** PIANO 4_LOTTAVE 49_NOTE **
1010 / -----
1020 /          di Gaetano Marano
1030 / -----
1040 /  (c) Gruppo Editoriale Jackson
1050 / -----
1060 MODE 1:INK 0,1:INK 1,26
1070 INK 2,11:INK 3,15
1080 PEN 1:PAPER 0:BORDER 1:CLS
1090 SOUND 1,0,0,0:SOUND 2,0,0,0
1100 SOUND 4,0,0,0
1110 / -----
1120 L#=CHR$(211):C#=CHR$(133):W#=CHR$(1
43):Z#=CHR$(209):D#=CHR$(138)
1130 PRINT " ";L#;L#;L#;C#;C#;W#;" PIAN
0 4_LOTTAVE 49_NOTE ";W#;D#;D#;Z#;Z#;Z#
1140 LOCATE 5,5
1150 PRINT "ATTENDERE UN'ATTIMO"
1160 DIM N(255)
1170 FOR A=0 TO 255
1180 N(A)=0:NEXT A
1190 RESTORE
1200 FOR A=1 TO 50
1210 READ K,D

```

```
1220 N(K)=D
1230 NEXT A
1240 LOCATE 5,5
1250 PRINT "
1260 K$=INKEY$
1270 IF K$<>" THEN K=ASC(K$)
1280 T=N(K)
1290 IF T<>0 THEN SOUND 1,T,30,14
1300 K=0
1310 GOTO 1260
1320 DATA 113,956,50,902,119,851
1330 DATA 51,804,101,758,114,716
1340 DATA 53,676,116,638,54,602
1350 DATA 121,568,55,536,117,506
1360 DATA 122,478,115,451,120,426
1370 DATA 100,402,99,379,118,358
1380 DATA 103,338,98,319,104,301
1390 DATA 110,284,106,268,109,253
1400 DATA 44,239,81,239,34,225
1410 DATA 87,213,35,201,69,190
1420 DATA 82,179,37,169,84,159
1430 DATA 38,150,89,142,39,134
1440 DATA 85,127,90,119,83,113
1450 DATA 88,106,68,100,67,95
1460 DATA 86,89,71,84,66,80,72,75
1470 DATA 78,71,74,67,77,63,60,60
```


PROGRAMMATORE DI SUONI

Una caratteristica interessante dei computer Amstrad è certamente il generatore di suoni a 3 voci.

Con il **Programmatore di Suoni** è possibile modificare facilmente e velocemente i valori che determinano le caratteristiche del suono fino ad arrivare al suono desiderato.

Usando i tasti di cursore a destra e a sinistra è possibile selezionare la **voce** da modificare, mentre tramite i tasti cursore in alto e in basso è possibile scegliere il parametro da modificare.

Poiché i valori massimi da inserire variano da 15 a 32767, è possibile scegliere anche il valore dell'incremento o del decremento tramite i tasti **U** (uno), **D** (dieci), **C** (cento) e **M** (mille); è importante tenere presente che se si cerca di decrementare un registro che si trova a 15 fino al valore di 8 dopo aver selezionato un decremento, ad esempio di cento, il decremento non avviene finché non si preme il tasto **U** per ritornare al decremento di uno.

Il parametro selezionato può essere incrementato o decrementato tramite i tasti + e - (da premere senza SHIFT) oltre a questi c'è il tasto **A** che consente di ascoltare il suono creato fino a quel momento e di avere anche le relative istruzioni **SOUND** ed il tasto **CLR** che azzerà tutto e fa ripartire il programma.

Da notare che all'inizio può sembrare che il programma non funzioni, infatti, per sentire ad esempio un suono dalla voce 1, occorre prima alzare il volume dell'altoparlante incorporato, quindi inserire una frequenza, una durata ed un volume per il suono e premere il tasto **A**.

Oltre ai parametri da inserire nelle istruzioni **SOUND** modifi-

cabili con questo programma, il generatore di suoni dispone anche delle istruzioni **ENT** ed **ENV** che consentono rispettivamente di scegliere l'involuppo di tono e di volume. Una descrizione più dettagliata del funzionamento e dell'uso delle istruzioni **SOUND**, **ENT** ed **ENV** è presente nel manuale dei computer **AMSTRAD**.

```

1000 /      ** PROGRAMMATORE DI SUONI **
1010 / -----
1020 /           di Gaetano Marano
1030 / -----
1040 /      (c) Gruppo Editoriale Jackson
1050 / -----
1060 MODE 1:INK 0,1:INK 1,26
1070 INK 2,11:INK 3,15
1080 PEN 1:PAPER 0:BORDER 1:CLS
1090 SOUND 1,0,0,0:SOUND 2,0,0,0
1100 SOUND 4,0,0,0
1110 / -----
1120 A#=CHR$(211):B#=CHR$(133):D#=CHR$(1
43):E#=CHR$(209):F#=CHR$(138)
1130 G#=CHR$(150):T#=CHR$(154):I#=CHR$(1
56):L#=CHR$(149):M#=CHR$(151):N#=CHR$(15
7):O#=CHR$(147):S#=CHR$(153):Q#=CHR$(154
)
1140 P#=CHR$(158):Z#=CHR$(155):X#=CHR$(1
59)
1150 R#=CHR$(154)
1160 T#=T#+T#+T#+T#+T#+T#+T#
1170 PRINT " ";A#;A#;A#;B#;B#;D#;"  PROG
RAMMATORE DI SUONI ";D#;F#;F#;E#;E#;E#
1180 W#=G#+T#+P#+T#+P#+T#+I#
1190 C#=M#+T#+X#+T#+X#+T#+N#
1200 Y#=O#+T#+Z#+T#+Z#+T#+S#
1210 LOCATE 16,4:PEN 0:PAPER 1:PRINT "VO
CE 1";:PEN 1:PAPER 0:PRINT "  VOCE 2  VO
CE 3"
1220 LOCATE 15,5:PRINT W#

```

```

1230 LOCATE 1,6:PRINT " TONO ----- ";L
$;" 0      ";L$;" 0      ";L$;" 0      ";L$
1240 LOCATE 15,7:PRINT C$
1250 LOCATE 1,8:PRINT " DURATA ----- ";L
$;" 0      ";L$;" 0      ";L$;" 0      ";L$
1260 LOCATE 15,9:PRINT C$
1270 LOCATE 1,10:PRINT " VOLUME ----- ";
L$;" 0      ";L$;" 0      ";L$;" 0      ";L
$
1280 LOCATE 15,11:PRINT C$
1290 LOCATE 1,12:PRINT " RUMORE ----- ";
L$;" 0      ";L$;" 0      ";L$;" 0      ";L
$
1300 LOCATE 15,13:PRINT Y$
1310 LOCATE 40,6:PRINT "<"
1320 DIM D(12,3)
1330 Y=6:I=1:V=1:C=16
1340 K$=INKEY$
1350 IF K$<>" " THEN K=ASC(K$)
1360 IF K=16 THEN RUN
1370 IF K=65 THEN GOSUB 1850
1380 IF K=242 OR K=243 THEN GOSUB 1730
1390 IF K=240 OR K=241 THEN GOSUB 1470
1400 IF K=85 OR K=117 THEN GOSUB 1690
1410 IF K=68 OR K=100 THEN GOSUB 1700
1420 IF K=67 OR K=99 THEN GOSUB 1710
1430 IF K=77 OR K=109 THEN GOSUB 1720
1440 IF K=59 OR K=45 THEN GOSUB 1550
1450 K=0:K$=""
1460 GOTO 1340
1470 H=Y
1480 IF K=241 THEN Y=Y+2
1490 IF K=240 THEN Y=Y-2
1500 IF Y=14 THEN Y=6
1510 IF Y=4 THEN Y=12
1520 LOCATE 40,H:PRINT " "
1530 LOCATE 40,Y:PRINT "<"

```

```

1540 RETURN
1550 IF Y=6 AND K=45 AND D(Y,V)-I>=0 THE
N D(Y,V)=D(Y,V)-I
1560 IF Y=6 AND K=59 AND D(Y,V)+I<4096 T
HEN D(Y,V)=D(Y,V)+I
1570 IF Y=8 AND K=45 AND D(Y,V)-I>=0 THE
N D(Y,V)=D(Y,V)-I
1580 IF Y=8 AND K=59 AND D(Y,V)+I<32768
THEN D(Y,V)=D(Y,V)+I
1590 IF Y=10 AND K=45 AND D(Y,V)-I>=0 TH
EN D(Y,V)=D(Y,V)-I
1600 IF Y=10 AND K=59 AND D(Y,V)+I<16 TH
EN D(Y,V)=D(Y,V)+I
1610 IF Y=12 AND K=45 AND D(Y,V)-I>=0 TH
EN D(Y,V)=D(Y,V)-I
1620 IF Y=12 AND K=59 AND D(Y,V)+I<16 TH
EN D(Y,V)=D(Y,V)+I
1630 GOSUB 1650
1640 RETURN
1650 LOCATE C,Y:PRINT "      "
1660 D$=STR$(D(Y,V))
1670 LOCATE C,Y:PRINT D$
1680 RETURN
1690 I=1:RETURN
1700 I=10:RETURN
1710 I=100:RETURN
1720 I=1000:RETURN
1730 IF K=243 THEN V=V+1
1740 IF K=242 THEN V=V-1
1750 IF V=0 THEN V=3
1760 IF V=4 THEN V=1
1770 LOCATE 16,4:PRINT "VOCE 1  VOCE 2
VOCE 3"
1780 C=V*8+8
1790 PEN 0:PAPER 1
1800 IF V=1 THEN LOCATE C,4:PRINT "VOCE
1"

```

```

1810 IF V=2 THEN, LOCATE C,4:PRINT "VOCE
2"
1820 IF V=3 THEN LOCATE C,4:PRINT "VOCE
3"
1830 PEN 1:PAPER 0
1840 RETURN
1850 FOR X=1 TO 3
1860 IF X=3 THEN X=4
1870 H$="SOUND "+RIGHT$(STR$(X),1)
1880 IF X=4 THEN X=3
1890 FOR L=6 TO 12 STEP 2
1900 J$=STR$(D(L,X))
1910 J$=RIGHT$(J$,LEN(J$)-1)
1920 H$=H$+" "+J$
1930 IF L=10 THEN H$=H$+"",0,0"
1940 NEXT L
1950 LOCATE 2,(16+X*2):PRINT H$;"
"
1960 NEXT X
1970 SOUND 1,D(6,1),D(8,1),D(10,1),0,0,D
(12,1)
1980 SOUND 2,D(6,2),D(8,2),D(10,2),0,0,D
(12,2)
1990 SOUND 4,D(6,3),D(8,3),D(10,3),0,0,D
(12,3)
2000 RETURN

```


SIMMETRICO

Questo è un programma di **animazione** che disegna sullo schermo delle linee orizzontali e verticali simmetricamente e contemporaneamente su quattro aree.

```
1000 /          \   *** SIMMETRICO ***
1010 / ----- \
1020 /           \   di Gaetano Marano
1030 / ----- \
1040 /   (c)   \   Gruppo Editoriale Jackson
1050 / ----- \
1060 MODE 0:INK 0,0:INK 1,26:INK 2,1
1070 INK 3,2:INK 4,4:INK 5,6:INK 6,7
1080 INK 7,9:INK 8,11:INK 9,15
1090 INK 10,16:INK 11,18:INK 12,24
1100 INK 13,23:INK 14,21:INK 15,5
1110 PEN 1:PAPER 0:BORDER 0:CLS
1120 SOUND 1,0,0,0:SOUND 2,0,0,0
1130 SOUND 4,0,0,0
1140 / ----- \
1150 RANDOMIZE TIME
1160 DX=60:DY=0
1170 D=INT(RND(1)*2)
1180 IF D=1 THEN W=DX:DX=DY:DY=W
1190 X=56+INT(RND(1)*265)
1200 Y=56+INT(RND(1)*145)
1210 SR=INT(RND(1)*4)
1220 C=1+INT(RND(1)*15)
1230 IF SR=0 THEN GOSUB 1260
1240 IF SR=1 OR SR=2 OR SR=3 THEN C=0:GO
SUB 1310
```

```
1250 GOTO 1170
1260 PLOT X,Y:DRAW X-DX,Y-DY,C
1270 PLOT 639-X,Y:DRAW 639-X+DX,Y-DY,C
1280 PLOT X,399-Y:DRAW X-DX,399-Y+DY,C
1290 PLOT 639-X,399-Y:DRAW 639-X+DX,399-
Y+DY,C
1300 RETURN
1310 PLOT X,Y:DRAW X-DX,Y-DY,C
1320 PLOT 639-X,Y:DRAW 639-X+DX,Y-DY,C
1330 PLOT X,399-Y:DRAW X-DX,399-Y+DY,C
1340 PLOT 639-X,399-Y:DRAW 639-X+DX,399-
Y+DY,C
1350 RETURN
```


TIRO ALLA FUNE

Il **tiro alla fune** è un gioco per due persone.

All'inizio appaiono sullo schermo due omini stilizzati con una fune.

L'omino di sinistra viene comandato dal tasto A mentre quello di destra dal tasto L. Ogni volta che appaiono le due stelle al centro dello schermo il giocatore che preme per primo il proprio tasto ha la possibilità di tirare i due omini dalla sua parte, se però uno dei giocatori preme il proprio tasto prima che appaiano le stelle, i due omini si sposteranno nel senso opposto favorendo l'avversario.

Il gioco continua finché uno dei due giocatori riesce a trascinare l'avversario completamente nella propria zona.

Il tempo di apparizione delle due stelle ed il ritardo tra una apparizione e l'altra sono casuali e vengono stabiliti rispettivamente dalle linee 1340 e 1280 che possono essere eventualmente modificate per rendere il gioco più difficile.

```
1000 /      *** TIRO ALLA FUNE ***
1010 / -----
1020 /      di Gaetano Marano
1030 / -----
1040 /      (c) Gruppo Editoriale Jackson
1050 / -----
1060 INK 0,0:INK 1,26:INK 2,11:INK 3,24
1070 BORDER 1:PAPER 0:PEN 1:CLS
1080 SOUND 1,0,0,0:SOUND 2,0,0,0
1090 SOUND 4,0,0,0
1100 / -----
1110 W#=CHR$(224):O#=CHR$(159):Z#=CHR$(1
```

```

55) :X#=CHR$(127):B#=CHR$(154):J#=CHR$(15
1):Y#=CHR$(157)
1120 Q#=CHR$(156):V#=CHR$(150):G#=CHR$(1
53):R#=CHR$(147):F#=CHR$(154):M#=CHR$(14
9)
1130 B#=B#+B#:B#=B#+B#:B#=B#+B#
1140 BB#=B#
1150 LOCATE 11,6:PEN 1:PRINT W#:LOCATE 3
0,6:PEN 2:PRINT W#
1160 LOCATE 11,7:PEN 1:PRINT Q#,Z#:B#:LO
CATE 21,7:PEN 2:PRINT BB#,Z#:Q#
1170 LOCATE 11,8:PEN 1:PRINT J#,Q#:LOCAT
E 29,8:PEN 2:PRINT V#,Y#
1180 LOCATE 11,9:PEN 1:PRINT R#,R#:LOCAT
E 29,9:PEN 2:PRINT G#,G#
1190 FOR T=1 TO 40
1200 LOCATE 0+T,10:PEN 3:PRINT X#
1210 NEXT T
1220 LOCATE 1,1:PEN 1:PRINT V#,F#,Q#:LOC
ATE 38,1:PEN 2:PRINT V#,F#,Q#
1230 LOCATE 1,2:PEN 1:PRINT M#;"A";M#:LO
CATE 38,2:PEN 2:PRINT M#;"L";M#
1240 LOCATE 1,3:PEN 1:PRINT R#,F#,G#:LOC
ATE 38,3:PEN 2:PRINT R#,F#,G#
1250 RANDOMIZE TIME:P=8
1260 M=-1
1270 LOCATE 20,4:PRINT "  "
1280 FOR A=1 TO 200+RND(1)*300
1290 K#=INKEY#
1300 IF K#<>" " THEN GOTO 1390
1310 NEXT A
1320 M=1
1330 LOCATE 20,4:PEN 3:PRINT "##"
1340 FOR A=1 TO 30+RND(1)*60
1350 K#=INKEY#
1360 IF K#<>" " THEN GOTO 1390
1370 NEXT A
1380 GOTO 1260

```

```

1390 IF K$="A" OR K$="a" THEN P=P-M
1400 IF K$="L" OR K$="l" THEN P=P+M
1410 LOCATE 1,6:PRINT "
"
1420 LOCATE P,6:PEN 1:PRINT W$:LOCATE P+
19,6:PEN 2:PRINT W$
1430 LOCATE 1,7:PRINT "
"
1440 LOCATE P,7:PEN 1:PRINT O#:Z#:B#:LOC
ATE P+10,7:PEN 2:PRINT BB#,Z#:O#
1450 LOCATE 1,8:PRINT "
"
1460 LOCATE P,8:PEN 1:PRINT J#:Q#:LOCATE
P+18,8:PEN 2:PRINT V#:Y#
1470 LOCATE 1,9:PRINT "
"
1480 LOCATE P,9:PEN 1:PRINT R#:R#:LOCATE
P+18,9:PEN 2:PRINT G#:G#
1490 IF P=1 OR P=21 THEN GOTO 1510
1500 GOTO 1260
1510 LOCATE 17,4
1520 IF P=1 THEN PEN 1:PRINT "Vince A"
1530 IF P=21 THEN PEN 2:PRINT "Vince L"

1540 LOCATE 10,18:PEN 1:PRINT "PREMERE G
PER GIOCARO"
1550 K#=INKEY#
1560 IF K#<>"G" AND K#<>"g" THEN GOTO 15
50
1570 RUN

```


ROULETTE

Il programma realizza una **Roulette** completa di sonoro. La richiesta di un numero avviene premendo il tasto **N**.

```
1000 /          *** ROULETTE ***
1010 / -----
1020 /          di Gaetano Marano
1030 / -----
1040 / (c) Gruppo Editoriale Jackson
1050 / -----
1060 MODE 1:INK 0,18:INK 1,0
1070 INK 2,6:INK 3,26
1080 PEN 3:PAPER 1:BORDER 1:CLS
1090 SOUND 1,0,0,0:SOUND 2,0,0,0
1100 SOUND 4,0,0,0
1110 / -----
1120 DIM P(37):DIM I(37):DIM NUM$(37)
1130 C=1:L=11:H=9
1140 FOR A=353 TO 0 STEP -9.729
1150 K=PI/180*A
1160 X=260*COS(K)+315
1170 Y=180*SIN(K)+210
1180 READ P,I,N$
1190 P(C)=P:I(C)=I:NUM$(C)=N$:C=C+1
1200 IF P=3 OR P=0 THEN GOSUB 1360
1210 X=X-1
1220 PEN I
1230 N=ASC(LEFT$(N$,1))-48
1240 GOSUB 1680
1250 X=X+15
1260 N=ASC(RIGHT$(N$,1))-48
```

```

1270 GOSUB 1680
1280 NEXT A
1290 PEN 3
1300 L=18:H=19
1310 R=50+INT(RND(1)*75)
1320 K#=INKEY#
1330 IF K#="N" OR K#="n" THEN GOSUB 1420
1340 PEN 3
1350 GOTO 1310
1360 FOR K=Y+(H/2) TO Y-H*2-(H/2) STEP -
1
1370 PEN P
1380 PLOT X-(L/2),K,P:DRAW X+L*2+(L/2),K
,P
1390 NEXT K
1400 PEN I
1410 RETURN
1420 V=0
1430 FOR F=6 TO R*9.729+6 STEP 9.729
1440 V=V+1:IF V=38 THEN V=1
1450 PLOT A,B,1
1460 PLOT A+1,B,1
1470 PLOT A,B+1,1
1480 PLOT A+1,B+1,1
1490 K=PI/180*F
1500 A=230*COS(K)+330
1510 B=200-150*SIN(K)
1520 PLOT A,B,3
1530 PLOT A+1,B,3
1540 PLOT A,B+1,3
1550 PLOT A+1,B+1,3
1560 FOR T=1 TO 30:NEXT T
1570 NEXT F
1580 X=270:Y=186:I=I(V):P=P(V)
1590 N#=NUM$(V)
1600 GOSUB 1360
1610 N=ASC(LEFT$(N#,1))-48

```

```

1620 GOSUB 1680
1630 X=X+22
1640 N=ASC(RIGHT$(N$,1))-48
1650 GOSUB 1680
1660 PEN 3
1670 RETURN
1680 IF N=-16 THEN RETURN
1690 IF N=0 OR N=2 OR N=3 OR N=5 OR N=6
OR N=7 OR N=8 OR N=9 THEN PLOT X,Y,I:DR
W X+L,Y,I
1700 IF N=2 OR N=3 OR N=4 OR N=5 OR N=6
OR N=8 OR N=9 THEN PLOT X,Y-H,I:DR
W X+L
,Y-H,I
1710 IF N=0 OR N=2 OR N=3 OR N=5 OR N=6
OR N=8 OR N=9 THEN PLOT X,Y-H-H,I:DR
W X
+L,Y-H-H,I
1720 IF N=0 OR N=1 OR N=2 OR N=3 OR N=4
OR N=7 OR N=8 OR N=9 THEN PLOT X+L,Y,I:DR
W X+L,Y-H,I
1730 IF N=0 OR N=1 OR N=3 OR N=4 OR N=5
OR N=6 OR N=7 OR N=8 OR N=9 THEN PLOT X+
L,Y-H,I:DR
W X+L,Y-H-H,I
1740 IF N=0 OR N=2 OR N=6 OR N=8 THEN PL
OT X,Y-H,I:DR
W X,Y-H-H,I
1750 IF N=0 OR N=4 OR N=5 OR N=6 OR N=8
OR N=9 THEN PLOT X,Y,I:DR
W X,Y-H,I
1760 RETURN
1770 DATA 1,2,"5 ",3,1,"24",1,2,"16",3,1
,"33",1,2,"1 "
1780 DATA 3,1,"20",1,2,"14",3,1,"31",1,2
,"9 ",3,1,"22"
1790 DATA 1,2,"18",3,1,"29",1,2," 7",3,1
,"28",1,2,"12"
1800 DATA 3,1,"35",1,2," 3",3,1,"26",0,1
," 0",1,2,"32"
1810 DATA 3,1,"15",1,2,"19",3,1," 4",1,2
,"21",3,1," 2"

```

```
1820 DATA 1,2,"25",3,1,"17",1,2,"34",3,1  
,"6 ",1,2,"27"  
1830 DATA 3,1,"13",1,2,"36",3,1,"11",1,2  
,"30";3,1,"8 "  
1840 DATA 1,2,"23",3,1,"10"
```


FLIPPER GIAPPONESE

In questo **Flipper Giapponese**, trenta palline cadono dall'alto attraverso una serie di chiodini, finendo in modo casuale su dei punteggi diversi, punteggi che vengono automaticamente sommati al totale.

```
1000 /      *** FLIPPER GIAPPONESE ***
1010 / -----
1020 /           di Gaetano Marano
1030 / -----
1040 /      (c) Gruppo Editoriale Jackson
1050 / -----
1060 MODE 1 INK 0,1:INK 1,26
1070 INK 2,6:INK 3,24
1080 PEN 1:PAPER 0:BORDER 1:CLS
1090 SOUND 1,0,0,0:SOUND 2,0,0,0
1100 SOUND 4,0,0,0
1110 / -----
1120 N$=CHR$(143)
1130 N$=N$+N$+N$+N$+N$+N$
1140 PEN 3
1150 LOCATE 14,2:PRINT N$;" ";N$
1160 A$=CHR$(155)+CHR$(32)
1170 A$=A$+A$+A$
1180 PEN 1
1190 FOR A=0 TO 12
1200 LOCATE 18-A,4+A:PRINT A$
1210 A$=A$+CHR$(155)+CHR$(32)
1220 NEXT A
1230 PAPER 3:PEN 0
```

```

1240 LOCATE 6,17:PRINT " 9 2 1
      1 2 9 "
1250 LOCATE 6,18:PRINT " 9 5 0 2 1
      1 2 0 5 9 "
1260 LOCATE 6,19:PRINT " 9 0 0 5 2 6 5 5
      6 2 5 0 0 9 "
1270 PEN 1:PAPER 0
1280 LOCATE 10,21:PRINT "PREMERE 6 PER P
ARTIRE "
1290 K#=INKEY#
1300 IF K#<>"6" AND K#<>"9" THEN GOTO 12
90
1310 LOCATE 10,21 PRINT "
      "
1320 PEN 0:PAPER 2:LOCATE 13,23:PRINT "P
UNTEGGIO          ":PEN 1:PAPER 0
1330 S=0:RANDOMIZE TIME
1340 F#=CHR$(231)
1350 FOR A=1 TO 30
1360 L=3:C=20
1370 LOCATE 2,3:PRINT A;" "
1380 PEN 2
1390 LOCATE 20,2:PRINT F#:FOR T=1 TO 450
:NEXT T:LOCATE 20,2:PRINT " ":LOCATE 20,
3:PRINT F#:LOCATE 20,3:PRINT " "
1400 FOR K=1 TO 13
1410 H=RND(1)
1420 IF H<0.5 THEN H=-1
1430 IF H>=0.5 THEN H=1
1440 LOCATE C,L:PRINT " "
1450 L=L+1
1460 LOCATE C+H,L:PEN 2:PRINT F#:PEN 1
1470 PRINT CHR$(7)
1480 C=C+H
1490 NEXT K
1500 IF C=19 OR C=21 THEN S=S+5
1510 IF C=17 OR C=23 THEN S=S+6
1520 IF C=15 OR C=25 THEN S=S+12

```

```
1530 IF C=13 OR C=27 THEN S=S+25
1540 IF C=11 OR C=29 THEN S=S+100
1550 IF C=9 OR C=31 THEN S=S+250
1560 IF C=7 OR C=33 THEN S=S+999
1570 G#=STR$(S):LOCATE 30,23:PRINT G#
1580 LOCATE C,L:PRINT " "
1590 NEXT A
1600 GOTO 1280
```


ROMBOSPIRALE

Altra animazione è **Rombospirale** che disegna e cancella alternativamente e in continuazione, una spirale a forma di rombo formata da caratteri e simboli scelti casualmente.

```
1000 /          *** ROMBOSPIRALE ***
1010 / -----
1020 /          di Gaetano Marano
1030 / -----
1040 /   (c) Gruppo Editoriale Jackson
1050 / -----
1060 MODE 1:INK 0,0:INK 1,11
1070 INK 2,6:INK 3,24
1080 PEN 1:PAPER 0:BORDER 1:CLS
1090 SOUND 1,0,0,0:SOUND 2,0,0,0
1100 SOUND 4,0,0,0
1110 / -----
1120 C=1
1130 GOSUB 1380
1140 FOR P=1 TO 11
1150 LOCATE P+10,14-P:PRINT CHR$(X)
1160 NEXT P
1170 FOR N=1 TO 11 STEP 2
1180 GOSUB 1380
1190 FOR P=1 TO 12-N
1200 LOCATE 20+P,1+P+N:PRINT CHR$(X)
1210 NEXT P
1220 GOSUB 1380
1230 FOR P=1 TO 12-N
1240 LOCATE 33-P-N,12+P:PRINT CHR$(X)
1250 NEXT P
```

```
1260 GOSUB 1380
1270 FOR P=1 TO 11-N
1280 LOCATE 22-P,25-P-N:PRINT CHR$(X)
1290 NEXT P
1300 GOSUB 1380
1310 FOR P=1 TO 11-N
1320 LOCATE 10+P+N,15-P:PRINT CHR$(X)
1330 NEXT P
1340 NEXT N
1350 C=C*(-1)
1360 X=32
1370 GOTO 1130
1380 IF C=1 THEN X=33+INT(RND(1)*222)
1390 PEN 1+INT(RND(1)*3)
1400 RETURN
```


Gli Amstrad presentano un ottimo rapporto prezzo/prestazioni e caratteristiche tecniche molto significative: grafica ad alta risoluzione, generatore di suoni, eccetera. Particolarmente interessante è il Locomotive BASIC che, oltre a essere tra i BASIC più completi, dispone anche di alcune innovazioni come le istruzioni AFTER e EVERY che offrono all'utente alcune forme di multitasking: il particolare l'eleborazione alternata di più parti del programma.

Le molte caratteristiche avanzate di questa macchina vengono sfruttate nei programmi presentati, vari tra loro: si va dalla grafica, alle utility, ai giochi. Programmi quindi pronti all'uso da analizzare, far girare, giocarci....

Elenco dei programmi:

- Istogrammi verticali
- Istogrammi orizzontali
- Cilindri 3D
- Design caratteri
- Disegnatore grafico
- Plotter
- Curve matematiche
- New York
- Disegni di linee
- Italia
- Numeri digitali
- Lettura veloce
- Cronometro
- Scritte pubblicitarie
- Contapezzi
- Fisica dei reattori nucleari
- Piano 4 ottave
- Programmatore di suoni
- Simmetrico
- Tiro alla fune
- Roulette
- Flipper giapponese
- Rombospirale



**Gaetano Marano
Progranni poer Amstard**

AMSTRAD

CPC



MÉMOIRE ÉCRITE
MEMORY ENGRAVED
MEMORIA ESCRITA



<https://acpc.me/>

[FRA] Ce document a été préservé numériquement à des fins éducatives et d'études, et non commerciales.

[ENG] This document has been digitally preserved for educational and study purposes, not for commercial purposes.

[ESP] Este documento se ha conservado digitalmente con fines educativos y de estudio, no con fines comerciales.